Script for the paper “*Uniform enclosures for the phase and zeros of Bessel functions and their derivatives*”

https://michaellevitin.net/bessels.html

Nikolay Filonov, Michael Levitin, Iosif Polterovich, and David A. Sher

---

Behind the scenes

- Directories

In[1]:=
```
curdir = SetDirectory[NotebookDirectory[]];
SaveDir = "./";
```

- Colors and symbols

In[3]:=
```
mycolors = ColorData[97, "ColorList"][[1 ;; 8]]
mytexcolors = "\\definecolor{mycolors1}{rgb}{0.368417, 0.506779,
    0.709798}\\definecolor{mycolors2}{rgb}{0.880722, 0.611041,
    0.142051}\\definecolor{mycolors3}{rgb}{0.560181, 0.691569, 0.194885}";
{filledcircle, filledsquare, filleddiamond, filleduptriangle, filleddowntriangle,
  circle, square, diamond, uptriangle, downtriangle} = Graphics`PlotMarkers[][[All, 1]]
```

Out[3]= {■, ■, ■, ■, ■, ■, ■, ■}

Out[5]= {●, ■, ◆, ▲, ▼, ○, □, ◇, △, ▽}

- MaTeX and lines

In[6]:=
```
Needs["MaTeX`"]
SetOptions[MaTeX, "BasePreamble" → {"\\usepackage{amsmath}",
    "\\usepackage{fourier}", "\\usepackage{ebgaramond}", "\\usepackage{bm}",
    "\\usepackage{xcolor}\\definecolor{darkgreen}{rgb}{0.00, 0.67, 0.00}"(*,
    "\\usepackage{accents}",
"\\DeclareMathAccent{\\wtilde}{\\mathord}{largesymbols}{\"65}",
"\\DeclareRobustCommand{\\utilde}[1]{\\underaccent{\\wtilde}{#1}}}"*)},
  FontSize → 12, Magnification → 1];
```

In[8]:=
```
Thk = AbsoluteThickness[1.8];
VeryThk = AbsoluteThickness[3.];
Thn = AbsoluteThickness[0.6];
res = 600; size = 4.00; sizesmall = 2.50; sizewide = 5.00;
hwcoeff = 1.2;
Imgsize = {UpTo[Round[72 size]], UpTo[Round[72 hwcoeff size]]};
Imgsizesmall = {UpTo[Round[72 sizesmall]], UpTo[Round[72 hwcoeff sizesmall]]};
Imgsizewide = {UpTo[Round[72 sizewide]], UpTo[Round[72 hwcoeff sizewide]]};
lightgr = GrayLevel[.9]; darkgr = GrayLevel[.6];
mydashing0 = Charting`ResolvePlotTheme["Monochrome", Plot]⟦7⟧⟦2⟧⟦5⟧⟦2⟧;
mydashing = Table[Directive[mydashing0⟦j, 2⟧, mydashing0⟦j, 4⟧], {j, 1, 8}];
mydsh = mydashing⟦2⟧; mydot = mydashing⟦3⟧;
PrTicks[xs_, ts_] := Thread[{xs, MaTeX[ts]}]
```

- mods for [Ho17]

In[21]:=
```
Mods[a_, b_] := a - b Round[a / b];
```

- findallRoots from https://mathematica.stackexchange.com/questions/16439/find-all-roots-of-an-interpolating-function-solution-to-a-differential-equation/16444#16444

In[22]:=
```
Clear[findAllRoots]
SyntaxInformation[findAllRoots] =
  {"LocalVariables" → {"Plot", {2, 2}}, "ArgumentsPattern" → {_, _, OptionsPattern[]}};
SetAttributes[findAllRoots, HoldAll];

Options[findAllRoots] = Join[{"ShowPlot" → False, PlotRange → All},
   FilterRules[Options[Plot], Except[PlotRange]]];

findAllRoots[fn_, {l_, lmin_, lmax_}, opts : OptionsPattern[]] :=
  Module[{pl, p, x, localFunction, brackets},
   localFunction = ReleaseHold[Hold[fn] /. HoldPattern[l] :→ x];
   If[lmin ≠ lmax, pl = Plot[localFunction, {x, lmin, lmax},
      Evaluate@FilterRules[Join[{opts}, Options[findAllRoots]], Options[Plot]]];
    p = Cases[pl, Line[{x__}] :→ x, Infinity];
    If[OptionValue["ShowPlot"],
     Print[Show[pl, PlotLabel → "Finding roots for this function",
       ImageSize → 200, BaseStyle → {FontSize → 8}]]], p = {}];
   brackets =
    Map[First, Select[(*This Split trick pretends that two points on the curve are
        "equal" if the function values have _opposite _ sign.Pairs of such sign-
       changes form the brackets for the subsequent FindRoot*)
      Split[p, Sign[Last[#2]] == -Sign[Last[#1]] &], Length[#1] == 2 &], {2}];
   x /. Apply[FindRoot[localFunction == 0, {x, ##1}] &, brackets, {1}] /. x → {}];
```

- cleanCountourPlot from https://mathematica.stackexchange.com/questions/3190/saner-alternative-to-contourplot-fill/3279#3279

In[27]:=
```
cleanContourPlot[cp_Graphics] := Module[{points, groups, regions, lines},
  groups = Cases[cp, {style__, g_GraphicsGroup} :> {{style}, g}, Infinity];
  points = First@Cases[cp, GraphicsComplex[pts_, ___] :> pts, Infinity];
  regions =
   Table[Module[{group, style, polys, edges, cover, graph}, {style, group} = g;
     polys = Join @@ Cases[group, Polygon[pt_, ___] :> pt, Infinity];
     edges = Join @@ (Partition[#, 2, 1, 1] & /@ polys);
     cover = Cases[Tally[Sort /@ edges], {e_, 1} :> e];
     graph = Graph[UndirectedEdge @@@ cover];
     {Sequence @@ style, FilledCurve[
       List /@ Line /@ First /@ Map[First, FindEulerianCycle /@ (Subgraph[graph, #] &) /@
            ConnectedComponents[graph], {3}]]}], {g, groups}];
  lines = Cases[cp, _Tooltip, Infinity];
  Graphics[GraphicsComplex[points, {regions, lines}], Sequence @@ Options[cp]]]

cleanContourPlot[Legended[cp_Graphics, rest___]] :=
 Legended[cleanContourPlot[cp], rest]
```

§1. Introduction and main results

§1.1. Setup I: Bessel functions

- Asymptotics of $\theta_\nu(x)$

In[29]:=
```
θasympt = x - Pi / 4 (2 ν + 1) + (4 ν^2 - 1) / (8 x) + (4 ν^2 - 1) (4 ν^2 - 25) / (384 x^3);
```

- Figure $C_{\nu,\tau}(x)$

```
In[30]:= xtib = {5, 10}; ytib = {-1, -0.5, 0, 0.5, 1};
figCnutau = GraphicsColumn[
  {GraphicsRow[Table[Plot[Cos[Pi τ] BesselJ[ν, x] + Sin[Pi τ] BesselY[ν, x] /.
        τ → {0, 1 / 4, 1 / 2, 3 / 4} // Evaluate, {x, 0, 10 + ν + ν^(1 / 3)},
      PlotStyle → mycolors⟦4 ;; 7⟧, AxesLabel → {MaTeX["x"], None},
      Ticks → {PrTicks[xtib, xtib], PrTicks[xtib, xtib]},
      Epilog → {Inset[Framed[MaTeX["\\nu=" <> ToString[ν]], RoundingRadius → 3],
         {Right, Bottom}, Scaled[{1.5, -0.25}]]}], {ν, {0, 3}}]],
   LineLegend[mycolors⟦4 ;; 7⟧,
    MaTeX[{"J_\\nu(x)", "\\mathcal{C}_{\\nu,\\frac{1}{4}}(x)", "Y_\\nu(x)",
      "\\mathcal{C}_{\\nu,\\frac{3}{4}}(x)"}], LegendLayout → "Row"]}]
```

Out[31]=




```
In[32]:= Export[SaveDir <> "figCnutau.pdf", figCnutau]
Out[32]=
./figCnutau.pdf
```

■ Computing $\theta_\nu(x)$ following [Ho17]

```
In[33]:= θ[ν_, x_] := If[Abs[x] ≥ Abs[ν], Sqrt[x^2 - ν^2] - ν ArcCos[ν / x] - Pi / 4, -Pi / 4] +
   Mods[ArcTan[BesselJ[ν, x], BesselY[ν, x]] -
     If[Abs[x] ≥ Abs[ν], Sqrt[x^2 - ν^2] - ν ArcCos[ν / x] - Pi / 4, -Pi / 4], 2 Pi];
```

■ Figure $\theta_\nu(x)$

```
In[34]:= yts1 = Table[yt, {yt, -Pi / 2, 3 Pi, Pi / 2}];
xts1 = Table[x, {x, 2, 10, 2}];
yti1 = PrTicks[yts1,
   {"-\\pi/2", "0", "\\pi/2", "\\pi", "3\\pi/2", "2\\pi", "5\\pi/2", "3\\pi"}];
xti1 = PrTicks[xts1, xts1];
```

In[38]:=
```
plotθ = Plot[Table[θ[ν, x], {ν, {0, 2, 5}}] // Evaluate, {x, 0, 10}, PlotStyle → mycolors,
  Ticks → {xti1, yti1}, AxesLabel → MaTeX[{"x", "\\theta_\\nu(x)"}], Epilog →
   {{Thin, Black, Dashed, Table[Line[{{0, yt}, {10, yt}}], {yt, 0, 3 Pi, Pi / 2}]},
    {mycolors[[1]], Dashed, Table[Line[
        {{BesselJZero[0, k], 0}, {BesselJZero[0, k], -Pi / 2 + Pi k}}], {k, 1, 3}], Table[
      Line[{{BesselYZero[0, k], 0}, {BesselYZero[0, k], -Pi + Pi k}}], {k, 1, 3}]},
    {mycolors[[1]], Table[
      Inset[filledcircle, {BesselJZero[0, k], 0}, Scaled[{1 / 2, 1 / 2}]], {k, 1, 3}]},
    {mycolors[[1]], Table[Inset[circle, {BesselYZero[0, k], 0}], {k, 1, 3}]},
    {mycolors[[2]], Dashed, Table[Line[
        {{BesselJZero[2, k], 0}, {BesselJZero[2, k], -Pi / 2 + Pi k}}], {k, 1, 2}], Table[
      Line[{{BesselYZero[2, k], 0}, {BesselYZero[2, k], -Pi + Pi k}}], {k, 1, 3}]},
    {mycolors[[2]], Table[Inset[filledcircle, {BesselJZero[2, k], 0}], {k, 1, 2}]},
    {mycolors[[2]], Table[Inset[circle, {BesselYZero[2, k], 0}], {k, 1, 3}]},
    {mycolors[[3]], Dashed, Table[Line[
        {{BesselJZero[5, k], 0}, {BesselJZero[5, k], -Pi / 2 + Pi k}}], {k, 1, 1}], Table[
      Line[{{BesselYZero[5, k], 0}, {BesselYZero[5, k], -Pi + Pi k}}], {k, 1, 2}]},
    {mycolors[[3]], Table[Inset[filledcircle, {BesselJZero[5, k], 0}], {k, 1, 1}]},
    {mycolors[[3]], Table[Inset[circle, {BesselYZero[5, k], 0}], {k, 1, 2}]}
   }, PlotLegends → MaTeX[{"\\nu=0", "\\nu=2", "\\nu=5"}]]
```

Out[38]=



In[39]:=
```
Export[SaveDir <> "plotthetabessels.pdf", plotθ]
```

Out[39]=
```
./plotthetabessels.pdf
```

■ Figure τ_ν^*

In[40]:=
```
figtaustar = Plot[1 / Pi θ[ν, ν] + 1 / 2, {ν, 0, 3},
  PlotRange → {0.0000, 0.18}, PlotStyle → {Thick, mycolors⟦4⟧},
  Epilog → {mycolors⟦4⟧, Thick, Line[{{0, 0}, {0, 0.042732809954193596}}]},
  (* some fiddling near ν==0*) AxesLabel → MaTeX[{"\\nu", "\\tau^*_\\nu"}],
  Ticks → {{Range[3], MaTeX[Range[3]]} // Transpose,
    {0.05 Range[3], MaTeX[0.05 Range[3]]} // Transpose}]
```

Out[40]=



In[41]:=
```
Export[SaveDir <> "figtaustar.pdf", figtaustar]
```

Out[41]=
```
./figtaustar.pdf
```

§1.2. Setup II: derivatives of Bessel functions

- Derivatives of Bessel functions

In[42]:=
```
DBesselJ[ν_, x_] := 1/2 (BesselJ[-1 + ν, x] - BesselJ[1 + ν, x]);
DBesselY[ν_, x_] := 1/2 (BesselY[-1 + ν, x] - BesselY[1 + ν, x]);
```

- Asymptotics of $\phi_\nu(x)$

In[44]:=
```
ϕasympt = x - Pi / 4 (2 ν - 1) + (4 ν^2 + 3) / (8 x) + (16 ν^4 + 184 ν^2 - 63) / (384 x^3);
```

- Computing $\phi_\nu(x)$ following [Ho17]

In[45]:=
```
ϕ[ν_, x_] := If[Abs[x] ≥ Abs[ν], Sqrt[x^2 - ν^2] - ν ArcCos[ν / x] + Pi / 4, Pi / 4] +
   Mods[ArcTan[DBesselJ[ν, x], DBesselY[ν, x]] -
     If[Abs[x] ≥ Abs[ν], Sqrt[x^2 - ν^2] - ν ArcCos[ν / x] + Pi / 4, Pi / 4], 2 Pi];
```

- Figure $C'_{\nu,\tau}(x)$

In[46]:=
```
xtib = {3, 6, 9, 12}; ytib = {-1, -0.5, 0, 0.5, 1};
figCnutauprime = GraphicsColumn[
  {GraphicsRow[Table[Plot[Cos[Pi τ] DBesselJ[ν, x] + Sin[Pi τ] DBesselY[ν, x] /.
       τ → {0, 1 / 4, 1 / 2, 9 / 10} // Evaluate, {x, 0, 10 + ν + ν ^ (1 / 3)},
      PlotStyle → mycolors[[4 ;; 7]], AxesLabel → {MaTeX["x"], None},
      Ticks → {PrTicks[xtib, xtib], PrTicks[ytib, ytib]},
      Epilog → {Inset[Framed[MaTeX["\\nu=" <> ToString[ν]], RoundingRadius → 3],
         {Right, Top}, Scaled[{1.5, 1.5}]]}], {ν, {0, 3}}]],
   LineLegend[mycolors[[4 ;; 7]],
    MaTeX[{"J'_\\nu(x)", "\\mathcal{C}'_{\\nu,\\frac{1}{4}}(x)", "Y'_\\nu(x)",
      "\\mathcal{C}'_{\\nu,\\frac{9}{10}}(x)"}], LegendLayout → "Row"]}]
```

Out[47]=



In[48]:=
```
Export[SaveDir <> "figCnutauprime.pdf", figCnutauprime]
```

Out[48]=

./figCnutauprime.pdf

■ Figure $\varphi_\nu(x)$

In[49]:=
```
plotϕjprimez = Table[findAllRoots[DBesselJ[ν, x], {x, 0, 10}], {ν, {0, 2, 5}}];
PrependTo[plotϕjprimez[[1]], 0];
plotϕjprimez
plotϕyprimez = Table[findAllRoots[DBesselY[ν, x], {x, 0, 10}], {ν, {0, 2, 5}}]
```

Out[51]=

{{0, 3.83171, 7.01559}, {3.05424, 6.70613, 9.96947}, {6.41562}}

Out[52]=

{{2.19714, 5.42968, 8.59601}, {5.00258, 8.35072}, {8.64956}}

In[53]:=
```
plotϕ = Plot[Table[ϕ[ν, x], {ν, {0, 2, 5}}] // Evaluate,
  {x, 0, 10}, PlotStyle → mycolors, Ticks → {xti1, yti1},
  AxesLabel → MaTeX[{"x", "\\varphi_\\nu(x)"}], AxesOrigin → {0, 0}, Epilog → {
    {Thin, Black, Dashed, Table[Line[{{0, yt}, {10, yt}}], {yt, 0, 3 Pi, Pi / 2}]},
    Table[
     {mycolors⟦j⟧,
      Dashed,
      Table[Line[{{plotϕjprimez⟦j, k⟧, 0}, {plotϕjprimez⟦j, k⟧, -Pi / 2 + Pi k}}],
       {k, 1, Length[plotϕjprimez⟦j⟧]}],
      Table[Line[{{plotϕyprimez⟦j, k⟧, 0}, {plotϕyprimez⟦j, k⟧, Pi k}}],
       {k, 1, Length[plotϕyprimez⟦j⟧]}]
     }, {j, 1, 3}],
    Table[
     {mycolors⟦j⟧,
      Table[Inset[filledcircle, {plotϕjprimez⟦j, k⟧, 0}],
       {k, 1, Length[plotϕjprimez⟦j⟧]}],
      Table[Inset[circle, {plotϕyprimez⟦j, k⟧, 0}], {k, 1, Length[plotϕyprimez⟦j⟧]}]
     }, {j, 1, 3}]
   }, PlotLegends → MaTeX[{"\\nu=0", "\\nu=2", "\\nu=5"}]]
```

Out[53]=



In[54]:=
```
Export[SaveDir <> "figphi.pdf", plotϕ]
```

Out[54]=
```
./figphi.pdf
```

§1.4. Definitions and properties of the auxiliary functions I

- $\tilde{\theta}_\nu(x)$, its derivative, and asymptotics

In[55]:=
```
θUp[ν_, x_] := Sqrt[x^2 - ν^2] - ν ArcCos[ν / x] - Pi / 4; (* θ̃ν(x) *)
DθUp = Simplify[D[θUp[ν, x], x], x > ν ≥ 0] (* θ̃ν′(x) *)
```

Out[56]=

$$\frac{\sqrt{x^2 - \nu^2}}{x}$$

In[57]:=
```
θUpasympt = Series[θUp[ν, x], {x, Infinity, 2}] // Simplify
```

Out[57]=

$$x - \frac{1}{4}\pi\,(1 + 2\,\nu) + \frac{\nu^2}{2\,x} + O\left[\frac{1}{x}\right]^3$$

- $\underset{\sim}{\theta}$(x), its derivative, and asymptotics

In[58]:=
```
θDown[ν_, x_] := Sqrt[x^2 - ν^2] - ν ArcCos[ν / x] -
   Pi / 4 - (3 x^2 + 2 ν^2) / (24 (x^2 - ν^2) ^ (3 / 2)); (* θ̰ν(x) *)
DθDown = Simplify[D[θDown[ν, x], x], x > ν ≥ 0]; (* θ̰′ν(x) *)
DθDown // TraditionalForm
Simplify[DθDown /. x → Sqrt[ν^2 + χ^2], χ > 0]
```

Out[60]//TraditionalForm=

$$\frac{-8\,\nu^6 + 8\,x^6 + \left(1 - 24\,\nu^2\right)x^4 + 4\left(6\,\nu^4 + \nu^2\right)x^2}{8\,x\left(x^2 - \nu^2\right)^{5/2}}$$

Out[61]=

$$\frac{5\,\nu^4 + 6\,\nu^2\,\chi^2 + \chi^4 + 8\,\chi^6}{8\,\chi^5\,\sqrt{\nu^2 + \chi^2}}$$

In[62]:=
```
θDownasympt = Series[θDown[ν, x], {x, Infinity, 4}] // Simplify
```

Out[62]=

$$x - \frac{1}{4}\pi\,(1 + 2\,\nu) + \frac{-1 + 4\,\nu^2}{8\,x} + \frac{\nu^2\left(-13 + 2\,\nu^2\right)}{48\,x^3} + O\left[\frac{1}{x}\right]^5$$

§1.5. Main results I: bounding the phase and zeros of Bessel functions

- Inverse functions of bounds with precision

In[63]:=
```
θUpInv[ν_, y_, prec_ : MachinePrecision] :=
  x /. FindRoot[θUp[ν, x] == y, {x, Max[ν, y + 1 / 4 Pi (2 ν + 1)], prec},
    WorkingPrecision → prec, Method → "AffineCovariantNewton"];
(* θDownInv[ν_,y_, prec_:MachinePrecision]:=
  x/.FindRoot[θDown[ν,x]==y,{x,Max[ν,y+1/4Pi (2ν+1)],prec}, WorkingPrecision→prec]//
   Re; *)
θDownInv[ν_, y_, prec_ : MachinePrecision] :=
  x /. FindRoot[θDown[ν, x] == y, {x, Max[ν, y + 1 / 4 Pi (2 ν + 1)], prec},
    WorkingPrecision → prec, Method → "AffineCovariantNewton"];
```

- Figure $\theta_\nu(x)$ bounds

In[65]:=
```
νs = {0, 5, 50, 200}; xmaxs = {3, 10, 58, 214};
xt = {{1, 2, 3}, {6, 8, 10}, {54, 58}, {207, 214}};
xticks = Table[{xt⟦j⟧, MaTeX[xt⟦j⟧, Magnification → 0.8]} // Transpose, {j, 1, 4}];
yticks = {{-1 / 2, 0, 1 / 4, 1 / 2}, MaTeX[{"-\\frac{1}{2}", "0",
      "\\frac{1}{4}", "\\frac{1}{2}"}, Magnification → 0.8]} // Transpose;
wp = 16;
jzeros = N[BesselJZero[νs, 1], wp]
jzerosdown = Table[θUpInv[νs⟦j⟧, Pi / 2, wp], {j, 1, 4}]
jzerosup = Table[θDownInv[νs⟦j⟧, Pi / 2, wp], {j, 1, 4}]
yzeros = N[BesselYZero[νs, 1], wp]
yzerosdown = Table[θUpInv[νs⟦j⟧, 0, wp], {j, 1, 4}]
yzerosup = Table[θDownInv[νs⟦j⟧, 0, wp], {j, 1, 4}]
jyzeros = Table[
  NSolveValues[{(BesselJ[ν, x] - BesselY[ν, x] /. ν → νs⟦j⟧) == 0, νs⟦j⟧ < x < xmaxs⟦j⟧},
    x, WorkingPrecision → wp]⟦1⟧, {j, 1, 4}]
jyzerosdown = Table[θUpInv[νs⟦j⟧, Pi / 4, wp], {j, 1, 4}]
jyzerosup = Table[θDownInv[νs⟦j⟧, Pi / 4, wp], {j, 1, 4}]
yzeros2 = N[BesselYZero[νs, 2], wp]
yzerosdown2 = Table[θUpInv[νs⟦j⟧, Pi, wp], {j, 1, 4}]
yzerosup2 = Table[θDownInv[νs⟦j⟧, Pi, wp], {j, 1, 4}]
```

Out[70]=
```
{2.404825557695773, 8.771483815959954, 57.11689916011917, 211.0291665105547}
```

Out[71]=
```
{2.356194490192345, 8.735670224368260, 57.06014735242129, 210.9435264448392}
```

Out[72]=
```
{2.408102577972088, 8.773723382042801, 57.12074576554989, 211.0349987107705}
```

Out[73]=
```
{0.8935769662791675, 6.747183824871022, 53.50285882040037, 205.4924725086642}
```

Out[74]=
```
{0.7853981633974483, 6.650743260039119, 53.32540833486605, 205.2177070708794}
```

Out[75]=
```
{0.9211047674809849, 6.773569788711935, 53.55298414169889, 205.5701650402530}
```

Out[76]=
```
{1.638559910293841, 7.799046601679099, 55.42736816079233, 208.4563055966140}
```

Out[77]=
```
{1.570796326794897, 7.746395558987345, 55.33814645332642, 208.3200599152652}
```

Out[78]=
```
{1.646705469222776, 7.805216821976846, 55.43844257964444, 208.4732660981063}
```

Out[79]=
```
{3.957678419314858, 10.59717672678203, 60.11244442774058, 215.5286305325874}
```

Out[80]=
```
{3.926990816987242, 10.57531070943339, 60.08131945448717, 215.4826177043710}
```

Out[81]=
```
{3.958567893185726, 10.59773018739399, 60.11333266661382, 215.5299550243314}
```

In[82]:=
```
xmaxs2 = {4.1, 11, 61, 217};
xt = {{2, 4}, {6, 8, 10}, {54, 58}, {208, 216}};
xticks = Table[{xt⟦j⟧, MaTeX[xt⟦j⟧, Magnification → 0.8]} // Transpose, {j, 1, 4}];
yticks = {yt = {-0.5, 0, 0.25, 0.5, 1}, MaTeX[yt, Magnification → 0.8],
    Table[1, 5], Table[Directive[Thin, Dashed], 5]} // Transpose;
figcomparisonJ2 = GraphicsGrid[
  ArrayReshape[Table[Plot[1 / Pi {θ[ν, x], θUp[ν, x], Max[-Pi / 2, θDown[ν, x] ]} /.
        ν → νs⟦j⟧ // Evaluate, {x, νs⟦j⟧, xmaxs2⟦j⟧},
      PlotLabel → MaTeX["\\nu=" <> ToString[νs⟦j⟧]], Epilog → {Black, Thin,
        Line[{{jzeros⟦j⟧, 0}, {jzeros⟦j⟧, 0.75}}],
        Line[{{yzeros⟦j⟧, 0}, {yzeros⟦j⟧, -0.6}}],
        Line[{{yzeros2⟦j⟧, 0}, {yzeros2⟦j⟧, 1.15}}],
        Line[{{jyzeros⟦j⟧, 0.25}, {jyzeros⟦j⟧, -0.25}}],
        Inset[MaTeX[ToString[DecimalForm[jzerosdown⟦j⟧, 6]] <>
            "<j_{\\nu,1}\\approx" <> ToString[DecimalForm[jzeros⟦j⟧, 6]] <>
            "<" <> ToString[DecimalForm[jzerosup⟦j⟧, 6]],
          Magnification → 0.8], {jzeros⟦j⟧, 0.75}, Scaled[{0.8, 0}]],
        Inset[MaTeX[ToString[DecimalForm[yzerosdown⟦j⟧, 6]] <>
            "<y_{\\nu,1}\\approx" <> ToString[DecimalForm[yzeros⟦j⟧, 6]] <>
            "<" <> ToString[DecimalForm[yzerosup⟦j⟧, 6]],
          Magnification → 0.8], {yzeros⟦j⟧, -0.6}, Scaled[{0, 1}]],
        Inset[MaTeX[ToString[DecimalForm[yzerosdown2⟦j⟧, 6]] <>
            "<y_{\\nu,2}\\approx" <> ToString[DecimalForm[yzeros2⟦j⟧, 6]] <>
            "<" <> ToString[DecimalForm[yzerosup2⟦j⟧, 6]],
          Magnification → 0.8], {yzeros2⟦j⟧, 1.1}, Scaled[{1, 0}]],
        Inset[MaTeX[ToString[DecimalForm[jyzerosdown⟦j⟧, 6]] <>
            "<c_{\\nu,\\frac{3}{4},1}\\approx" <> ToString[DecimalForm[
              jyzeros⟦j⟧, 6]] <> "<" <> ToString[DecimalForm[jyzerosup⟦j⟧, 6]],
          Magnification → 0.8], {jyzeros⟦j⟧, -0.25}, Scaled[{0.2, 1}]]
       }, PlotRange → {-0.75, 1.25}, Ticks → {xticks⟦j⟧, yticks}], {j, 1, 4}], {2, 2}],
  Frame → All]
```

Out[86]=



In[87]:=
```
Export["fig_comparison_J.pdf", figcomparisonJ2]
```

Out[87]=
```
fig_comparison_J.pdf
```

§1.6. Definitions and properties of the auxiliary functions II

■ $\underset{\sim}{\phi}_\nu(x)$, its derivative, and asymptotics

In[88]:=
```
ϕDown[ν_, x_] := Sqrt[x^2 - ν^2] - ν ArcCos[ν / x] + Pi / 4;  (* ϕ̰ν(x) *)
DϕDown = Simplify[D[ϕDown[ν, x], x], x > ν ≥ 0] (* ϕ̰′ν(x) *)
```

Out[89]=

$$\frac{\sqrt{x^2-\nu^2}}{x}$$

In[90]:=
```
ϕDownasympt = Series[ϕDown[ν, x], {x, Infinity, 2}] // Simplify
```

Out[90]=

$$x+\frac{1}{4}(\pi-2\pi\nu)+\frac{\nu^2}{2x}+O\left[\frac{1}{x}\right]^3$$

■ $\tilde{\phi}_\nu(x)$, its derivative, and asymptotics

In[91]:=
```
ϕUp[ν_, x_] :=
  Sqrt[x^2 - ν^2] - ν ArcCos[ν / x] + Pi / 4 + (9 x^2 - 2 ν^2) / (24 (x^2 - ν^2) ^ (3 / 2));
DϕUp = Simplify[D[ϕUp[ν, x], x], x > ν ≥ 0]
```

Out[92]=

$$\frac{8x^6-8\nu^6+4x^2\nu^2\left(-1+6\nu^2\right)-3x^4\left(1+8\nu^2\right)}{8x\left(x^2-\nu^2\right)^{5/2}}$$

In[93]:=
```
ϕUpasympt = Series[ϕUp[ν, x], {x, Infinity, 3}] // Simplify
```

Out[93]=

$$x + \frac{1}{4}(\pi - 2\pi\nu) + \frac{3 + 4\nu^2}{8x} + \frac{\nu^2(23 + 2\nu^2)}{48x^3} + O\left[\frac{1}{x}\right]^4$$

- p_ν and $z_\nu^\star$

In[94]:=
```
pν[ν_, x_] := 8 x^6 - 8 ν^6 + 4 x^2 ν^2 (-1 + 6 ν^2) - 3 x^4 (1 + 8 ν^2);
```

In[95]:=
```
xstar[ν_] :=
  If[ν == 0, Sqrt[3 / 8], Root[-8 ν^6 + (-4 ν^2 + 24 ν^4) #1^2 + (-3 - 24 ν^2) #1^4 + 8 #1^6 &, 2]];
zstar[ν_] := ϕUp[ν, xstar[ν]]
```

In[97]:=
```
Assuming[ ν > 0, Series[xstar[ν], {ν, Infinity, 1}]]
Assuming[ν > 0, Series[ϕUp[ν, ν + 1/4 × 7^(1/3) ν^(1/3)], {ν, Infinity, 1}]] // FullSimplify
```

Root: Because of branch cuts, the series may represent a different root of
$-8 + (24\nu^2 - 4\nu^4)\#1^2 + (-24\nu^4 - 3\nu^6)\#1^4 + 8\nu^6\#1^6$ & for some values of ν.

Out[97]=

$$\nu + \frac{1}{4} \times 7^{1/3}\nu^{1/3} + \frac{19\left(\frac{1}{\nu}\right)^{1/3}}{96 \times 7^{1/3}} + \frac{5}{384\nu} + O\left[\frac{1}{\nu}\right]^{4/3}$$

Out[98]=

$$\frac{1}{12}\left(2\sqrt{14} + 3\pi\right) + \frac{\sqrt{2}\left(\frac{1}{\nu}\right)^{2/3}}{5 \times 7^{1/6}} + O\left[\frac{1}{\nu}\right]^{4/3}$$

In[99]:=
```
zstar[0] / Pi // N
zstarinf = Pi / 4 + Sqrt[7 / 18];
zstarinf / Pi // N
```

Out[99]=

```
0.639848
```

Out[101]=

```
0.448501
```

In[102]:=

```
ztzt = {ztz = {0.5, 0.6}, MaTeX[ztz]} // Transpose;
PrependTo[ztzt, {N[zstarinf / Pi], MaTeX["0.4485"], 1}];
AppendTo[ztzt, {N[zstar[0] / Pi], MaTeX["0.6398"]}];
figzstar = Plot[zstar[ν] / Pi, {ν, 0, 10}, PlotRange → {Full, Full}, AxesLabel → MaTeX[
    {"\\nu", "\\frac{z^\\star_\\nu}{\\pi}"}], AxesOrigin → {0, zstarinf / Pi - 0.02},
  Ticks → {{xtz = {3, 6, 9}, MaTeX[xtz]} // Transpose, ztzt}, AspectRatio → 1 / 2]
```

Out[104]=



In[105]:=

```
Export[SaveDir <> "figzstar.pdf", figzstar]
```

Out[105]=

```
./figzstar.pdf
```

§1.7. Main results II: bounding zeros of derivatives of Bessel functions

- Figure “Forbidden region”

In[106]:=

```
zstar[0] / Pi - 1 / 2 // N
FindRoot[zstar[ν] / Pi == 1 / 2, {ν, 1}]
figforbid = RegionPlot [τ ≤ zstar[ν] / Pi - 1 / 2, {ν, 0, 1.4},
  {τ, 0, 0.15}, Frame → Off, Axes → True, AxesLabel → MaTeX[{"\\nu", "\\tau"}],
  AspectRatio → 1 / 2, PlotStyle → Directive[Opacity[0.25], mycolors⟦4⟧],
  BoundaryStyle → mycolors⟦4⟧, Ticks → {{xtr = {1, 1.19876}, MaTeX[xtr]} // Transpose,
    {ytr = {0.1, 0.1398}, MaTeX[ytr]} // Transpose}]
```

Out[106]=

```
0.139848
```

Out[107]=

```
{ν → 1.19876}
```

Out[108]=



In[109]:=

```
Export[SaveDir <> "figforbid.pdf", figforbid]
```

Out[109]=

```
./figforbid.pdf
```

■ Inverse functions of bounds with precision

In[110]:=

```
ϕUpInv[ν_, y_, prec_ : MachinePrecision] :=
  x /. FindRoot[ϕUp[ν, x] == y, {x, Max[ν, y + 1 / 4 Pi (2 ν + 1)], prec},
    WorkingPrecision → prec, Method → "AffineCovariantNewton"];
ϕDownInv[ν_, y_, prec_ : MachinePrecision] :=
  x /. FindRoot[ϕDown[ν, x] == y, {x, Max[ν, y + 1 / 4 Pi (2 ν + 1)], prec},
    WorkingPrecision → prec, Method → "AffineCovariantNewton"];
wp = 8;
```

■ Figure $\phi_\nu(x)$ bounds

In[113]:=

```
xmaxs2 = {4.1, 11, 61, 217};
jpzeros = Table[NSolveValues[{(DBesselJ[ν, x] /. ν → νs⟦j⟧) == 0, νs⟦j⟧ < x < xmaxs2⟦j⟧},
    x, WorkingPrecision → wp], {j, 1, 4}];
PrependTo[jpzeros⟦1⟧, 0];
jpzeros
jpzerosdown =
 Table[If[j == 1 && k == 0, , ϕUpInv[νs⟦j⟧, Pi / 2 + Pi k, wp]], {j, 1, 4}, {k, 0, 1}]
jpzerosup =
 Table[If[j == 1 && k == 0, , ϕDownInv[νs⟦j⟧, Pi / 2 + Pi k, wp]], {j, 1, 4}, {k, 0, 1}]
ypzeros =
 Table[If[j > 1, NSolveValues[{(DBesselY[ν, x] /. ν → νs⟦j⟧) == 0, νs⟦j⟧ < x < xmaxs2⟦j⟧},
     x, WorkingPrecision → wp]⟦1⟧, N[BesselYZero[1, 1], wp]], {j, 1, 4}]
ypzerosdown = Table[ϕUpInv[νs⟦j⟧, Pi, wp], {j, 1, 4}]
ypzerosup = Table[ϕDownInv[νs⟦j⟧, Pi, wp], {j, 1, 4}]
jypzeros = Table[
  NSolveValues[{(DBesselJ[ν, x] - DBesselY[ν, x] /. ν → νs⟦j⟧) == 0, Floor[ypzeros⟦j⟧] <
      x < Ceiling[jpzeros⟦j⟧⟦2⟧]}, x, WorkingPrecision → wp]⟦1⟧, {j, 1, 4}]
jypzerosdown = Table[NSolveValues[
    {ϕUp[νs⟦j⟧, x] == 5 Pi / 4, Floor[ypzeros⟦j⟧] < x < Ceiling[jpzeros⟦j⟧⟦2⟧]},
    x, WorkingPrecision → wp]⟦1⟧, {j, 1, 4}]
jypzerosup = Table[NSolveValues[
    {ϕDown[νs⟦j⟧, x] == 5 Pi / 4, Floor[ypzeros⟦j⟧] < x < Ceiling[jpzeros⟦j⟧⟦2⟧]},
    x, WorkingPrecision → wp]⟦1⟧, {j, 1, 4}]
```

Out[116]=

```
{{0, 3.8317060}, {6.4156164, 10.519861}, {52.997640, 60.026319}, {204.74096, 215.41064}}
```

Out[117]=

```
{{Null, 3.8290554}, {6.2882622, 10.518842},
 {52.818788, 60.024996}, {204.47655, 215.40874}}
```

Out[118]=

```
{{Null, 3.9269908}, {6.6507433, 10.575311},
 {53.325410, 60.081320}, {205.21772, 215.48262}}
```

Out[119]=

```
{2.1971413, 8.6495562, 56.962904, 210.81029}
```

Out[120]=

```
{2.1845331, 8.6452479, 56.956742, 210.80123}
```

Out[121]=

```
{2.3561945, 8.7356702, 57.060152, 210.94355}
```

Out[122]=

```
{3.0224970, 9.6046632, 58.552684, 213.20737}
```

Out[123]=

```
{3.0173098, 9.6027478, 58.550077, 213.20358}
```

Out[124]=

```
{3.1415927, 9.6717325, 58.623174, 213.30163}
```

In[125]:=

```
yticks = {yt = {0, 0.5, 1, 1.25, 1.5}, MaTeX[yt, Magnification → 0.8],
    Table[1, 5], Table[Directive[Thin, Dashed], 5]} // Transpose;
xt = {{2, 4}, {7, 10}, {54, 58}, {208, 216}};
xticks = Table[{xt⟦j⟧, MaTeX[xt⟦j⟧, Magnification → 0.8]} // Transpose, {j, 1, 4}];
Do[PrependTo[xticks⟦j⟧,
  {xstar[νs⟦j⟧], MaTeX["x^\\star_\\nu", Magnification → 0.8], 0}], {j, 1, 4}]
figcomparisonJp = GraphicsGrid[ArrayReshape[Table[
    Show[
     Plot[1 / Pi {ϕ[ν, x], ϕUp[ν, x], ϕDown[ν, x]} /. ν → νs⟦j⟧ // Evaluate,
      {x, xstar[νs⟦j⟧], xmaxs2⟦j⟧},
      PlotLabel → MaTeX["\\nu=" <> ToString[νs⟦j⟧]], Epilog → {Black, Thin,
        Line[{{jpzeros⟦j⟧⟦2⟧, 0}, {jpzeros⟦j⟧⟦2⟧, 1.65}}],
        Line[{{jypzeros⟦j⟧, 1.25}, {jypzeros⟦j⟧, -0.45}}],
        Line[{{ypzeros⟦j⟧, 0}, {ypzeros⟦j⟧, 1.3}}],
        If[j > 1, Line[{{jpzeros⟦j⟧⟦1⟧, 0.5}, {jpzeros⟦j⟧⟦1⟧, -0.25}}]],
        Line[{{xstar[νs⟦j⟧], 0}, {xstar[νs⟦j⟧], zstar[νs⟦j⟧] / Pi}}],
        Inset[MaTeX[ToString[DecimalForm[jpzerosdown⟦j⟧⟦2⟧, 6]] <>
           "<j'_{\\nu,2}\\approx" <> ToString[DecimalForm[jpzeros⟦j⟧⟦2⟧, 6]] <>
           "<" <> ToString[DecimalForm[jpzerosup⟦j⟧⟦2⟧, 6]],
          Magnification → 0.8], {Right, 1.65}, Scaled[{1, 0}]],
        If[j > 1, Inset[MaTeX[ToString[DecimalForm[jpzerosdown⟦j⟧⟦1⟧, 6]] <>
            "<j'_{\\nu,1}\\approx" <> ToString[DecimalForm[jpzeros⟦j⟧⟦1⟧, 6]] <>
            "<" <> ToString[DecimalForm[jpzerosup⟦j⟧⟦1⟧, 6]],
           Magnification → 0.8], {νs⟦j⟧, -0.25}, Scaled[{0, 1}]]],
        Inset[MaTeX[ToString[DecimalForm[ypzerosdown⟦j⟧, 6]] <>
           "<y'_{\\nu,1}\\approx" <> ToString[DecimalForm[ypzeros⟦j⟧, 6]] <>
           "<" <> ToString[DecimalForm[ypzerosup⟦j⟧, 6]],
          Magnification → 0.8], {ypzeros⟦j⟧, 1.3}, Scaled[{0.8, 0}]],
        Inset[MaTeX[ToString[DecimalForm[jypzerosdown⟦j⟧, 6]] <>
           "<c'_{\\nu,\\frac{3}{4},1}\\approx" <> ToString[DecimalForm[
              jypzeros⟦j⟧, 6]] <> "<" <> ToString[DecimalForm[jypzerosup⟦j⟧, 6]],
          Magnification → 0.8], {Right, -0.45}, Scaled[{1, 1}]]
       },
      PlotRange → {{νs⟦j⟧, xmaxs2⟦j⟧}, {-0.7, 1.85}}, Ticks → {xticks⟦j⟧, yticks}],
     Plot[1 / Pi {ϕ[ν, x], zstar[ν], ϕDown[ν, x]} /. ν → νs⟦j⟧ // Evaluate,
      {x, νs⟦j⟧, xstar[νs⟦j⟧]}, PlotRange → {Automatic, {-0.7, 1.85}}]
    ], {j, 1, 4}], {2, 2}], Frame → All]
```

Out[129]=



In[130]:=

```
Export[SaveDir <> "fig_comparison_Jp.pdf", figcomparisonJp]
```

Out[130]=

```
./fig_comparison_Jp.pdf
```

§2. Proofs

§2.1. Liouville’s transform and phase functions

In[131]:=

```
𝓕[f_, t_, x_] := Cos[f - Pi t] / Sqrt[D[f, x]];
𝓥[f_, x_] :=
 D[f, x] ^2 + 1 / 2 D[f, {x, 3}] / D[f, x] - 3 / 4 (D[f, {x, 2}] / D[f, x]) ^2 // Simplify
CheckSchrö[𝓐_, 𝓟_, x_] := D[𝓐, x, x] + 𝓟 𝓐 // Simplify;
(* checks Schrödinger equation for function 𝓐 and potential 𝓟 *)
```

■ Lemma 2.1

In[134]:=

```
CheckSchrö[𝓕[f[x], t, x], 𝓥[f[x], x], x]
```

Out[134]=

```
0
```

■ Lemma 2.3

In[135]:=
```
Simplify[D[ArcTan[𝒜2[x] / 𝒜1[x]], x]] /.
 Wronskian[{𝒜1[x], 𝒜2[x]}, x] → 𝒲 (*𝒜1[x] 𝒜2′[x]-𝒜2[x] 𝒜1′[x]→𝒲*)
```
Out[135]=
$$\frac{\mathcal{W}}{\mathcal{A}1[x]^2 + \mathcal{A}2[x]^2}$$

In[136]:=
```
Simplify[
 FullSimplify[ℱ[ArcTan[𝒜2[x] / 𝒜1[x]], t, x] // TrigExpand, 𝒜1[x] > 0 && 𝒲 > 0] /.
  Wronskian[{𝒜1[x], 𝒜2[x]}, x] → 𝒲, 𝒜1[x]^2 + 𝒜2[x]^2 > 0]
```
Out[136]=
$$\frac{\mathrm{Cos}[\pi\, t]\, \mathcal{A}1[x] + \mathrm{Sin}[\pi\, t]\, \mathcal{A}2[x]}{\sqrt{\mathcal{W}}}$$

In[137]:=
```
(* extra check *)
Simplify[𝒱[ArcTan[𝒜2[x] / 𝒜1[x]], x] /.
  {𝒜1''[x] → -𝒫[x] × 𝒜1[x], 𝒜1'''[x] → -𝒫'[x] × 𝒜1[x] - 𝒫[x] × 𝒜1'[x],
   𝒜2''[x] → -𝒫[x] × 𝒜2[x], 𝒜2'''[x] → -𝒫'[x] × 𝒜2[x] - 𝒫[x] × 𝒜2'[x]}]
```
Out[137]=
```
𝒫[x]
```

- Lemma 2.4

In[138]:=
```
CheckSchrö[Sqrt[x] BesselJ[ν, x], 1 - (ν^2 - 1 / 4) / x^2, x] // FullSimplify
```
Out[138]=
```
0
```

In[139]:=
```
𝒱θ[ν_, x_] := 1 - (ν^2 - 1 / 4) / x^2;
```

In[140]:=
```
CheckSchrö[x^(3 / 2) / Sqrt[x^2 - ν^2] DBesselJ[ν, x],
  1 - (ν^2 - 1 / 4) / x^2 - (2 ν^2 + x^2) / (x^2 - ν^2)^2, x] // FullSimplify
```
Out[140]=
```
0
```

In[141]:=
```
𝒱ϕ[ν_, x_] := 1 - (ν^2 - 1 / 4) / x^2 - (2 ν^2 + x^2) / (x^2 - ν^2)^2;
```

§2.3. Proof of Theorem 1.4

- Upper bound

In[142]:=
```
Map[Collect[#, x] &, Simplify[𝒱[θUp[ν, x], x], x > ν ≥ 0]]
```
Out[142]=
$$\frac{4\, x^6 - 12\, x^4\, \nu^2 + \nu^4 - 4\, \nu^6 + 6\, x^2\, \nu^2\, \left(-1 + 2\, \nu^2\right)}{4\, \left(x^3 - x\, \nu^2\right)^2}$$

In[143]:=

$$\mathcal{V}\theta\text{Up}[\nu\_, x\_] := \frac{4 x^6 - 12 x^4 \nu^2 + \nu^4 - 4 \nu^6 + 6 x^2 \nu^2 \left(-1 + 2 \nu^2\right)}{4 \left(x^3 - x \nu^2\right)^2};$$

- Condition (C_2)

In[144]:=

```
Simplify[𝒱θ[ν, x] - 𝒱θUp[ν, x], x > ν ≥ 0]
```

Out[144]=

$$\frac{x^2 + 4 \nu^2}{4 \left(x^2 - \nu^2\right)^2}$$

- Condition $\left(C_3'\right)$

In[145]:=

```
θUpasympt - θasympt // Simplify
```

Out[145]=

$$\frac{1}{8 x} + O\left[\frac{1}{x}\right]^3$$

- Lower bound

In[146]:=

```
{q1, q2} = Simplify[𝒱[θDown[ν, x], x], x > ν ≥ 0] // Together // NumeratorDenominator;
```

In[147]:=

```
Collect[q1, x, Simplify] // TraditionalForm
```

Out[147]//TraditionalForm=

$$1024 \nu^{22} (4 \nu^2 - 1) + 4096 x^{24} + 2048 (1 - 24 \nu^2) x^{22} + 128 (2112 \nu^4 - 128 \nu^2 + 15) x^{20} + 32 (-28160 \nu^6 + 1760 \nu^4 + 584 \nu^2 + 1) x^{18} + (2027520 \nu^8 - 107520 \nu^6 - 117376 \nu^4 + 704 \nu^2 + 1) x^{16} - 16 \nu^2 (202752 \nu^8 - 7680 \nu^6 - 13088 \nu^4 + 223 \nu^2 - 1) x^{14} + 16 \nu^4 (236544 \nu^8 - 5376 \nu^6 - 5000 \nu^4 + 585 \nu^2 + 6) x^{12} - 16 \nu^6 (202752 \nu^8 - 2688 \nu^6 + 11440 \nu^4 + 935 \nu^2 - 16) x^{10} + 16 \nu^8 (126720 \nu^8 - 1920 \nu^6 + 15272 \nu^4 + 823 \nu^2 + 16) x^8 - 128 \nu^{12} (7040 \nu^6 - 240 \nu^4 + 808 \nu^2 + 43) x^6 + 256 \nu^{14} (1056 \nu^6 - 80 \nu^4 + 17 \nu^2 + 3) x^4 - 1024 \nu^{18} (48 \nu^4 - 7 \nu^2 - 5) x^2$$

In[148]:=

$$\mathcal{V}\theta\text{Down}[\nu\_, x\_] := \left(x^{16} + 32 x^{18} + 1920 x^{20} + 2048 x^{22} + 4096 x^{24} + 16 x^{14} \nu^2 + 704 x^{16} \nu^2 + 18688 x^{18} \nu^2 - 16384 x^{20} \nu^2 - 49152 x^{22} \nu^2 + 96 x^{12} \nu^4 - 3568 x^{14} \nu^4 - 117376 x^{16} \nu^4 + 56320 x^{18} \nu^4 + 270336 x^{20} \nu^4 + 256 x^{10} \nu^6 + 9360 x^{12} \nu^6 + 209408 x^{14} \nu^6 - 107520 x^{16} \nu^6 - 901120 x^{18} \nu^6 + 256 x^8 \nu^8 - 14960 x^{10} \nu^8 - 80000 x^{12} \nu^8 + 122880 x^{14} \nu^8 + 2027520 x^{16} \nu^8 + 13168 x^8 \nu^{10} - 183040 x^{10} \nu^{10} - 86016 x^{12} \nu^{10} - 3244032 x^{14} \nu^{10} - 5504 x^6 \nu^{12} + 244352 x^8 \nu^{12} + 43008 x^{10} \nu^{12} + 3784704 x^{12} \nu^{12} + 768 x^4 \nu^{14} - 103424 x^6 \nu^{14} - 30720 x^8 \nu^{14} - 3244032 x^{10} \nu^{14} + 4352 x^4 \nu^{16} + 30720 x^6 \nu^{16} + 2027520 x^8 \nu^{16} + 5120 x^2 \nu^{18} - 20480 x^4 \nu^{18} - 901120 x^6 \nu^{18} + 7168 x^2 \nu^{20} + 270336 x^4 \nu^{20} - 1024 \nu^{22} - 49152 x^2 \nu^{22} + 4096 \nu^{24}\right) / \left(64 x^2 \left(x^2 - \nu^2\right)^5 \left(x^4 + 8 x^6 + 4 x^2 \nu^2 - 24 x^4 \nu^2 + 24 x^2 \nu^4 - 8 \nu^6\right)^2\right);$$

- Condition (C_2)

In[149]:=

```
Collect[FullSimplify[(64 χ^5 (5 ν^4 + 6 ν^2 χ + χ^2 + 8 χ^3)^2)
    (𝒱θDown[ν, Sqrt[ν^2 + χ]] - 𝒱θ[ν, Sqrt[ν^2 + χ]])], χ, Simplify] // TraditionalForm
```

Out[149]//TraditionalForm=

$$625 \nu^{14} + 2375 \nu^{12} \chi + 3525 \nu^{10} \chi^2 + (33984 \nu^2 + 16) \chi^8 + 5 (400 \nu^2 + 519) \nu^8 \chi^3 + (5200 \nu^2 + 1011) \nu^6 \chi^4 + (99008 \nu^4 + 784 \nu^2 + 1) \chi^7 + (70720 \nu^4 + 1696 \nu^2 + 23) \nu^2 \chi^6 + 3 (1376 \nu^2 + 71) \nu^4 \chi^5 + 1600 \chi^9$$

- Condition $\left(C_3'\right)$

In[150]:=

```
θasympt - θDownasympt // Simplify
```

Out[150]=

$$\frac{25}{384\,x^3} + O\left[\frac{1}{x}\right]^5$$

§2.4 Proof of Lemma 1.8

In[151]:=

```
DϕUpN = Collect[Simplify[pν[ν, Sqrt[ν^2 + ξ]], ξ > 0], ξ, Simplify]
DerϕUpN = D[DϕUpN, ξ]
Dχ0 = SolveValues[DerϕUpN == 0 && ξ > 0,  ξ, Assumptions → ν ≥ 0] // Apart
```

Out[151]=

$$-7\,\nu^4 - 10\,\nu^2\,\xi - 3\,\xi^2 + 8\,\xi^3$$

Out[152]=

$$-10\,\nu^2 - 6\,\xi + 24\,\xi^2$$

Out[153]=

$$\left\{\frac{1}{8} + \frac{\sqrt{3 + 80\,\nu^2}}{8\,\sqrt{3}}\right\}$$

In[154]:=

```
dnu = FullSimplify[((D[ϕUp[μ, x], μ] // Simplify)) /. x → κ μ, κ > 1 && μ ≥ 0]
```

Out[154]=

$$-\frac{2 - 23\,\kappa^2 + 24\,\left(-1 + \kappa^2\right)^{5/2}\,\mu^2\,\text{ArcSec}[\kappa]}{24\,\left(-1 + \kappa^2\right)^{5/2}\,\mu^2}$$

In[155]:=

```
μsn = Simplify[SolveValues[dnu == 0 , μ, Assumptions → κ > 1][[1]], κ > 1 ]
```

Out[155]=

$$\frac{\sqrt{\frac{-2 + 23\,\kappa^2}{\text{ArcSec}[\kappa]}}}{2\,\sqrt{6}\,\left(-1 + \kappa^2\right)^{5/4}}$$

In[156]:=

```
κeq = Simplify[pν[μ, κ μ] /. μ → μsn, κ > 1 ]
```

Out[156]=

$$-\frac{\left(2 - 23\,\kappa^2\right)^2\,\left(-2 + 25\,\kappa^2 - 23\,\kappa^4 + 3\,\kappa^2\,\sqrt{-1 + \kappa^2}\,\left(4 + 3\,\kappa^2\right)\,\text{ArcSec}[\kappa]\right)}{1728\,\left(-1 + \kappa^2\right)^{11/2}\,\text{ArcSec}[\kappa]^3}$$

In[157]:=

```
Simplify[SolveValues[(κeq /. ArcSec[κ] → A) == 0, A], κ > 1 ][[1]] - ArcSec[κ]
```

Out[157]=

$$\frac{2 - 25\,\kappa^2 + 23\,\kappa^4}{3\,\kappa^2\,\sqrt{-1 + \kappa^2}\,\left(4 + 3\,\kappa^2\right)} - \text{ArcSec}[\kappa]$$

In[158]:=

$$K[\varkappa\_] := \frac{2 - 25\,\varkappa^2 + 23\,\varkappa^4}{3\,\varkappa^2\,\sqrt{-1+\varkappa^2}\,\left(4 + 3\,\varkappa^2\right)} - \mathrm{ArcSec}[\varkappa];$$

```
Simplify[K'[κ], κ > 1]
```

Out[159]=

$$-\frac{16\,\left(-1+\varkappa^2\right)^{3/2}\,\left(1+6\,\varkappa^2\right)}{3\,\varkappa^3\,\left(4+3\,\varkappa^2\right)^2}$$

§. Proof of Theorem 1.9

- Lower bound

In[160]:=

```
𝒱ϕDown[ν_, x_] := 𝒱θUp[ν, x];
```

- Condition (C_2)

In[161]:=

```
Simplify[𝒱ϕDown[ν, x] - 𝒱ϕ[ν, x], x > ν ≥ 0]
```

Out[161]=

$$\frac{3\,x^2 + 4\,\nu^2}{4\,\left(x^2-\nu^2\right)^2}$$

- Condition (C_3')

In[162]:=

```
ϕasympt - ϕDownasympt // Simplify
```

Out[162]=

$$\frac{3}{8\,x} + O\left[\frac{1}{x}\right]^3$$

- Upper bound

In[163]:=

```
{q3, q4} = Simplify[𝒱[ϕUp[ν, x], x], x > ν ≥ 0] // Together // NumeratorDenominator
```

Out[163]=

$$\{81\,x^{16} - 864\,x^{18} - 1152\,x^{20} - 6144\,x^{22} + 4096\,x^{24} + 432\,x^{14}\,\nu^2 + 1344\,x^{16}\,\nu^2 - 24\,320\,x^{18}\,\nu^2 + 40\,960\,x^{20}\,\nu^2 - 49\,152\,x^{22}\,\nu^2 + 864\,x^{12}\,\nu^4 - 9200\,x^{14}\,\nu^4 + 138\,624\,x^{16}\,\nu^4 - 91\,136\,x^{18}\,\nu^4 + 270\,336\,x^{20}\,\nu^4 + 768\,x^{10}\,\nu^6 + 26\,512\,x^{12}\,\nu^6 - 261\,632\,x^{14}\,\nu^6 - 9216\,x^{16}\,\nu^6 - 901\,120\,x^{18}\,\nu^6 + 256\,x^8\,\nu^8 - 25\,584\,x^{10}\,\nu^8 + 165\,760\,x^{12}\,\nu^8 + 466\,944\,x^{14}\,\nu^8 + 2\,027\,520\,x^{16}\,\nu^8 + 7664\,x^8\,\nu^{10} + 91\,392\,x^{10}\,\nu^{10} - 1\,118\,208\,x^{12}\,\nu^{10} - 3\,244\,032\,x^{14}\,\nu^{10} - 640\,x^6\,\nu^{12} - 183\,680\,x^8\,\nu^{12} + 1\,419\,264\,x^{10}\,\nu^{12} + 3\,784\,704\,x^{12}\,\nu^{12} + 768\,x^4\,\nu^{14} + 80\,896\,x^6\,\nu^{14} - 1\,112\,064\,x^8\,\nu^{14} - 3\,244\,032\,x^{10}\,\nu^{14} - 768\,x^4\,\nu^{16} + 546\,816\,x^6\,\nu^{16} + 2\,027\,520\,x^8\,\nu^{16} - 5120\,x^2\,\nu^{18} - 159\,744\,x^4\,\nu^{18} - 901\,120\,x^6\,\nu^{18} + 23\,552\,x^2\,\nu^{20} + 270\,336\,x^4\,\nu^{20} - 1024\,\nu^{22} - 49\,152\,x^2\,\nu^{22} + 4096\,\nu^{24}, 64\,x^2\,\left(x^2-\nu^2\right)^5\,\left(-3\,x^4 + 8\,x^6 - 4\,x^2\,\nu^2 - 24\,x^4\,\nu^2 + 24\,x^2\,\nu^4 - 8\,\nu^6\right)^2\}$$

In[164]:=

```
Collect[q3, x, Simplify] // TraditionalForm
```

Out[164]//TraditionalForm=

$$1024\,\nu^{22}\left(4\,\nu^2-1\right)+4096\,x^{24}-6144\left(8\,\nu^2+1\right)x^{22}+128\left(2112\,\nu^4+320\,\nu^2-9\right)x^{20}-32\left(28\,160\,\nu^6+2848\,\nu^4+760\,\nu^2+27\right)x^{18}+3\left(675\,840\,\nu^8-3072\,\nu^6+46\,208\,\nu^4+448\,\nu^2+27\right)x^{16}-16\,\nu^2\left(202\,752\,\nu^8-29\,184\,\nu^6+16\,352\,\nu^4+575\,\nu^2-27\right)x^{14}+16\,\nu^4\left(236\,544\,\nu^8-69\,888\,\nu^6+10\,360\,\nu^4+1657\,\nu^2+54\right)x^{12}-48\,\nu^6\left(67\,584\,\nu^8-29\,568\,\nu^6-1904\,\nu^4+533\,\nu^2-16\right)x^{10}+16\,\nu^8\left(126\,720\,\nu^8-69\,504\,\nu^6-11\,480\,\nu^4+479\,\nu^2+16\right)x^8-128\,\nu^{12}\left(7040\,\nu^6-4272\,\nu^4-632\,\nu^2+5\right)x^6+768\,\nu^{14}\left(352\,\nu^6-208\,\nu^4-\nu^2+1\right)x^4-1024\,\nu^{18}\left(48\,\nu^4-23\,\nu^2+5\right)x^2$$

In[165]:=

```
𝒱ϕUp[ν_, x_] :=
  (81 x^16 - 864 x^18 - 1152 x^20 - 6144 x^22 + 4096 x^24 + 432 x^14 ν^2 + 1344 x^16 ν^2 - 24 320 x^18 ν^2 +
     40 960 x^20 ν^2 - 49 152 x^22 ν^2 + 864 x^12 ν^4 - 9200 x^14 ν^4 + 138 624 x^16 ν^4 - 91 136 x^18 ν^4 +
     270 336 x^20 ν^4 + 768 x^10 ν^6 + 26 512 x^12 ν^6 - 261 632 x^14 ν^6 - 9216 x^16 ν^6 - 901 120 x^18 ν^6 +
     256 x^8 ν^8 - 25 584 x^10 ν^8 + 165 760 x^12 ν^8 + 466 944 x^14 ν^8 + 2 027 520 x^16 ν^8 + 7664 x^8 ν^10 +
     91 392 x^10 ν^10 - 1 118 208 x^12 ν^10 - 3 244 032 x^14 ν^10 - 640 x^6 ν^12 - 183 680 x^8 ν^12 +
     1 419 264 x^10 ν^12 + 3 784 704 x^12 ν^12 + 768 x^4 ν^14 + 80 896 x^6 ν^14 - 1 112 064 x^8 ν^14 -
     3 244 032 x^10 ν^14 - 768 x^4 ν^16 + 546 816 x^6 ν^16 + 2 027 520 x^8 ν^16 - 5120 x^2 ν^18 - 159 744 x^4 ν^18 -
     901 120 x^6 ν^18 + 23 552 x^2 ν^20 + 270 336 x^4 ν^20 - 1024 ν^22 - 49 152 x^2 ν^22 + 4096 ν^24) /
   (64 x^2 (x^2 - ν^2)^5 (-3 x^4 + 8 x^6 - 4 x^2 ν^2 - 24 x^4 ν^2 + 24 x^2 ν^4 - 8 ν^6)^2);
```

- Condition (C_2)

In[166]:=

```
Simplify[4096 x^2 (x^2 - ν^2)^10 D[ϕUp[ν, x], x]^2 (𝒱ϕ[ν, x] - 𝒱ϕUp[ν, x]), x > ν ≥ 0]
```

Out[166]=

$$4032\,x^{18}+4096\,\nu^{18}+1024\,x^2\,\nu^{14}\left(-1+17\,\nu^2\right)-3584\,x^4\,\nu^{12}\left(-1+41\,\nu^2\right)+16\,x^{16}\left(27+1208\,\nu^2\right)-64\,x^8\,\nu^6\left(12-337\,\nu^2+2338\,\nu^4\right)+64\,x^6\,\nu^8\left(-4-164\,\nu^2+4543\,\nu^4\right)-16\,x^{10}\,\nu^4\left(54+1415\,\nu^2+12\,796\,\nu^4\right)+16\,x^{12}\,\nu^2\left(-27+701\,\nu^2+20\,272\,\nu^4\right)-x^{14}\left(81+2640\,\nu^2+158\,656\,\nu^4\right)$$

In[167]:=

```
δ𝒱ϕ[ν_, x_] := 4032 x^18 + 4096 ν^18 + 1024 x^2 ν^14 (-1 + 17 ν^2) -
   3584 x^4 ν^12 (-1 + 41 ν^2) + 16 x^16 (27 + 1208 ν^2) - 64 x^8 ν^6 (12 - 337 ν^2 + 2338 ν^4) +
   64 x^6 ν^8 (-4 - 164 ν^2 + 4543 ν^4) - 16 x^10 ν^4 (54 + 1415 ν^2 + 12 796 ν^4) +
   16 x^12 ν^2 (-27 + 701 ν^2 + 20 272 ν^4) - x^14 (81 + 2640 ν^2 + 158 656 ν^4);
```

In[168]:=

```
(δ𝒱ϕ[0, x] x^(-14) // Simplify) /. x → Sqrt[3 / 8 + ξ] // Simplify
```

Out[168]=

$$72\left(9+48\,\xi+56\,\xi^2\right)$$

In[169]:=

```
δ𝒱ϕ1 = Collect[ν^(-14) δ𝒱ϕ[ν, ν + ν ξ], ν, Simplify]
```

Out[169]=

$$-(1+\xi)^6\left(7+6\,\xi+3\,\xi^2\right)^4+16\,\xi^3\,(1+\xi)^2\,(2+\xi)^3\left(343+1120\,\xi+1672\,\xi^2+1304\,\xi^3+998\,\xi^4+1008\,\xi^5+672\,\xi^6+216\,\xi^7+27\,\xi^8\right)\nu^2+64\,\xi^6\,(2+\xi)^6\left(1463+4410\,\xi+5681\,\xi^2+3980\,\xi^3+1625\,\xi^4+378\,\xi^5+63\,\xi^6\right)\nu^4$$

In[170]:=

```
δineq1 = δ𝒱ϕ1 /. ν^2 → 0
```

Out[170]=

$$-(1+\zeta)^6\left(7+6\zeta+3\zeta^2\right)^4+64\zeta^6(2+\zeta)^6\left(1463+4410\zeta+5681\zeta^2+3980\zeta^3+1625\zeta^4+378\zeta^5+63\zeta^6\right)\nu^4$$

In[171]:=

```
Collect[ν^-4 pν[ν, ν (ζ + 1)] // Simplify, ν]
νineq = FullSimplify[Reduce[{ν^-4 pν[ν, ν (ζ + 1)] ≥ 0, ζ > 0, ν > 0}, ν], ζ > 0]
Map[FullSimplify[#^4, ζ > 0] &, νineq]
```

Out[171]=

$$-4(1+\zeta)^2-3(1+\zeta)^4+\left(-8+24(1+\zeta)^2-24(1+\zeta)^4+8(1+\zeta)^6\right)\nu^2$$

Out[172]=

$$\nu \geq \frac{(1+\zeta)\sqrt{7+3\zeta(2+\zeta)}}{2\sqrt{2}(\zeta(2+\zeta))^{3/2}}$$

Out[173]=

$$\nu^4 \geq \frac{(1+\zeta)^4(7+3\zeta(2+\zeta))^2}{64\zeta^6(2+\zeta)^6}$$

In[174]:=

```
Collect[δineq1 /. ν → νineq[[2]], ζ, Simplify] // FullSimplify
```

Out[174]=

$$2(1+\zeta)^4(7+3\zeta(2+\zeta))^2(707+\zeta(2+\zeta)(1057+\zeta(2+\zeta)(409+27\zeta(2+\zeta))))$$

- Condition (C_3')

In[175]:=

```
ϕUpasympt - ϕasympt // Simplify
```

Out[175]=

$$\frac{21}{128x^3}+O\left[\frac{1}{x}\right]^4$$

§3. Derivatives of ultraspherical Bessel functions

§3.1. Setup III

- Definitions

In[176]:=

```
D[x^(-η) BesselJ[ν, x], x] // FullSimplify
```

Out[176]=

$$x^{-\eta}\left(\text{BesselJ}[-1+\nu,x]-\frac{(\eta+\nu)\,\text{BesselJ}[\nu,x]}{x}\right)$$

In[177]:=

```
Uprime[ν_, η_, x_] := x^-η (BesselJ[-1 + ν, x] - ((η + ν) BesselJ[ν, x])/x)
Wprime[ν_, η_, x_] := x^-η (BesselY[-1 + ν, x] - ((η + ν) BesselY[ν, x])/x)
```

- Figure $U'_{\nu,\eta}(x)$ and $W'_{\nu,\eta}(x)$

In[179]:=

```
xtib = {5, 10}; ytib = {-10, -5, 0, 5, 10};
figUprimeWprime =
 GraphicsColumn[{GraphicsRow[Table[Plot[{x^2 Uprime[ν, 1, x], x^11 Uprime[ν, 10, x],
        x^2 Wprime[ν, 1, x], x^11 Wprime[ν, 10, x]} // Evaluate,
      {x, 0, 10 + ν + ν^(1/3)}, PlotStyle → mycolors⟦4 ;; 7⟧, AxesLabel →
       {MaTeX["x"], None}, Ticks → {PrTicks[xtib, xtib], PrTicks[ytib, ytib]},
      Epilog → {Inset[Framed[MaTeX["\\nu=" <> ToString[ν]], RoundingRadius → 3],
         {Right, If[ν == 0, Bottom, Top]},
         Scaled[{1.5, If[ν == 0, -0.25, 1.25]}]]}], {ν, {0, 3}}]],
   LineLegend[mycolors⟦4 ;; 7⟧,
    MaTeX[{"x^2 U'_{\\nu,1}(x)", "x^{11} U'_{\\nu,10}(x)",
      "x^2 W'_{\\nu,1}(x)", "x^{11} W'_{\\nu,10}(x)"}], LegendLayout → "Row"]}]
```

Out[180]=



In[181]:=

```
Export[SaveDir <> "figUWprime.pdf", figUprimeWprime]
```

Out[181]=

```
./figUWprime.pdf
```

§3.2. Phase function of ultraspherical Bessel derivatives

- Lemma 3.1

In[182]:=

```
μ2[ν_, η_] := ν^2 - η^2;
𝒱ψ[ν_, η_, x_] :=
  1 - (ν^2 - 1/4) / x^2 + 2 (1 - η) / (x^2 - μ2[ν, η]) - 3 x^2 / (x^2 - μ2[ν, η])^2;
```

In[184]:=

```
CheckSchrö[x^(η+3/2)/Sqrt[x^2-μ2[ν, η]] Uprime[ν, η, x], 𝒱ψ[ν, η, x], x] //
 FullSimplify
CheckSchrö[x^(η+3/2)/Sqrt[x^2-μ2[ν, η]] Wprime[ν, η, x], 𝒱ψ[ν, η, x], x] //
 FullSimplify
```

Out[184]=

0

Out[185]=

0

In[186]:=

```
Wronskian[{x^(η+3/2)/Sqrt[x^2-μ2[ν, η]] Uprime[ν, η, x],
  x^(η+3/2)/Sqrt[x^2-μ2[ν, η]] Wprime[ν, η, x]}, x]
```

Out[186]=

$$\frac{2}{\pi}$$

In[187]:=

```
D[ArcTan[Wprime[ν, η, x]/Uprime[ν, η, x]], x] - 2 (x^2 - μ2[ν, η]) /
    (Pi x^(2 η + 3) (Uprime[ν, η, x]^2 + Wprime[ν, η, x]^2)) // FullSimplify
```

Out[187]=

0

- Lemma 3.2

In[188]:=

```
seriesLsquared = Collect[Asymptotic[
    x^(2 η) (Uprime[ν, η, x]^2 + Wprime[ν, η, x]^2), {x, Infinity, 5}] // Simplify, x]
```

Out[188]=

$$\frac{2}{\pi x} + \frac{3 + 8\eta + 8\eta^2 - 4\nu^2}{4\pi x^3} + \frac{\left(45 + 48\eta + 16\eta^2 - 4\nu^2\right)\left(-1 + 4\nu^2\right)}{64\pi x^5}$$

In[189]:=

```
ser = Series[2/Pi (x^2 - ν^2 + η^2)/x^3 / seriesLsquared, {x, Infinity, 4}] // Normal;
seras = Collect[Integrate[ser, x] + C, x,
  Collect[Numerator[#], {ν, η}, Simplify]/Denominator[#] &]
```

Out[190]=

$$C + x + \frac{3 + 8\eta + 4\nu^2}{8x} + \frac{-63 - 144\eta - 192\eta^2 - 128\eta^3 + (184 + 192\eta)\nu^2 + 16\nu^4}{384 x^3}$$

In[191]:=

```
x^(-η) Asymptotic[Sqrt[seriesLsquared] Cos[seras], {x, Infinity, 1}]
Asymptotic[Uprime[ν, η, x], {x, Infinity, 1}]
```

Out[191]=

$$\sqrt{\frac{2}{\pi}}\, x^{-\frac{1}{2}-\eta} \operatorname{Cos}\left[C + x + \frac{3 + 8\eta + 4\nu^2}{8x}\right]$$

Out[192]=

$$\sqrt{\frac{2}{\pi}}\, x^{-\frac{1}{2}-\eta} \operatorname{Cos}\left[x - \frac{1}{4}\pi(-1 + 2\nu)\right] + \frac{x^{-\frac{3}{2}-\eta}\left(1 - 4(-1+\nu)^2\right) \operatorname{Sin}\left[x - \frac{1}{4}\pi(-1 + 2\nu)\right]}{4\sqrt{2\pi}}$$

In[193]:=

```
ψasympt = seras /. C → -1/4 π (-1 + 2 ν)
```

Out[193]=

$$x - \frac{1}{4}\pi(-1+2\nu) + \frac{3+8\eta+4\nu^2}{8x} + \frac{-63-144\eta-192\eta^2-128\eta^3+(184+192\eta)\nu^2+16\nu^4}{384x^3}$$

§3.3. Definitions and properties of the auxiliary functions III

§3.4. Main results III: bounding the phase and zeros of derivatives of ultraspherical Bessel functions

- $\underset{\sim}{\psi}_{\nu,\eta}(x)$ and its derivative

In[194]:=

```
ψDown[ν_, η_, x_] := Sqrt[x^2 - μ^2] + η / Sqrt[x^2 - μ^2] +
   Pi / 4 (η^2 / μ + 2 (μ - ν) + 1) - (μ + η^2 / (2 μ)) ArcCos[μ / x];
Collect[Limit[ψDown[ν, η, x] /. μ → Sqrt[μ2[ν, η]],
  η → ν, Direction → "FromBelow", Assumptions → x > 0], x]
ψDown1[ν_, η_, x_] := If[ν > η, Sqrt[x^2 - ν^2 + η^2] +
    η / Sqrt[x^2 - ν^2 + η^2] + Pi / 4 (η^2 / Sqrt[ν^2 - η^2] + 2 (Sqrt[ν^2 - η^2] - ν) + 1) -
    (Sqrt[ν^2 - η^2] + η^2 / (2 Sqrt[ν^2 - η^2])) ArcCos[Sqrt[ν^2 - η^2] / x],
   x + ν (2 + ν)/(2 x) + 1/4 (π - 2 π ν)
  ];
```

Out[195]=

$$x + \frac{\nu(2+\nu)}{2x} + \frac{1}{4}(\pi - 2\pi\nu)$$

In[197]:=

```
DψDown = FullSimplify[D[ψDown[ν, η, x], x] // Together, x > μ > 0]
```

Out[197]=

$$\frac{2x^4 + \mu^2(\eta^2 + 2\mu^2) - x^2(\eta(2+\eta) + 4\mu^2)}{2x((x-\mu)(x+\mu))^{3/2}}$$

- $x^{\sharp}_{\mu,\eta}$

In[198]:=

```
FullSimplify[SolveValues[Numerator[DψDown] == 0, x][[4]], η > 0]
```

Out[198]=

$$\frac{1}{2}\sqrt{\eta(2+\eta) + 4\mu^2 + \sqrt{\eta(\eta(2+\eta)^2 + 16\mu^2)}}$$

In[199]:=

```
xhash[ν_, η_] := 1/2 Sqrt[η (2 + η) + 4 μ2[ν, η] + Sqrt[η (η (2 + η)^2 + 16 μ2[ν, η])]]
xhashμ[μ_, η_] := 1/2 Sqrt[η (2 + η) + 4 μ^2 + Sqrt[η (η (2 + η)^2 + 16 μ^2)]]
```

- $\mathcal{V}_{\underset{\sim}{\psi}_{\nu,\eta}}$

In[201]:=

```
𝒱ψDown = 𝒱[ψDown[ν, η, x], x];
{q5, q6} = Simplify[𝒱ψDown // Together // NumeratorDenominator, x > μ > 0];
(* q5=Collect[q5, x, Collect[#,η,Collect[#,μ,Simplify]&]&] *)
q5 = Collect[q5, {x}, Simplify];
q6 = FullSimplify[Expand[q6], x > 0];
q5 // TraditionalForm
q6 // TraditionalForm
```

Out[205]//TraditionalForm=

$$\mu^8 \left(\eta^2+2\mu^2\right)^2 \left(\eta^4+4\eta^2\mu^2+4\mu^4-\mu^2\right)+16x^{16}-32x^{14}\left(\eta^2+2\eta+4\mu^2\right)+$$
$$8x^{12}\left(3\eta^4+12\eta^3+\eta^2\left(28\mu^2+9\right)+6\eta\left(8\mu^2-1\right)+\mu^2\left(56\mu^2-3\right)\right)-$$
$$4x^{10}\left(2\eta^6+12\eta^5+12\eta^4\left(3\mu^2+2\right)+8\eta^3\left(15\mu^2+2\right)+14\eta^2\mu^2\left(12\mu^2+5\right)+12\eta\mu^2\left(20\mu^2-1\right)+\mu^4\left(224\mu^2-31\right)\right)+$$
$$x^8\left(\eta^8+8\eta^7+8\eta^6\left(5\mu^2+3\right)+32\eta^5\left(6\mu^2+1\right)+2\eta^4\left(180\mu^4+145\mu^2+8\right)+\right.$$
$$\left.48\eta^3\mu^2\left(20\mu^2+3\right)+4\eta^2\mu^2\left(280\mu^4+99\mu^2+6\right)+8\eta\mu^4\left(160\mu^2+13\right)+20\mu^6\left(56\mu^2-13\right)\right)-$$
$$\mu^2x^6\left(4\eta^8+24\eta^7+16\eta^6\left(5\mu^2+3\right)+32\eta^5\left(9\mu^2+1\right)+\eta^4\mu^2\left(480\mu^2+293\right)+\eta^3\left(960\mu^4+76\mu^2\right)+\right.$$
$$\left.4\eta^2\mu^2\left(280\mu^4+56\mu^2+9\right)+120\eta\left(8\mu^6+\mu^4\right)+56\mu^6\left(16\mu^2-5\right)\right)+$$
$$\mu^4x^4\left(6\eta^8+24\eta^7+8\eta^6\left(10\mu^2+3\right)+192\eta^5\mu^2+9\eta^4\mu^2\left(40\mu^2+11\right)+24\eta^3\mu^2\left(20\mu^2-1\right)+\right.$$
$$\left.4\eta^2\mu^2\left(168\mu^4+4\mu^2+3\right)+24\eta\mu^4\left(16\mu^2-1\right)+32\mu^6\left(14\mu^2-5\right)\right)-\mu^6x^2\left(4\eta^8+8\eta^7+40\eta^6\mu^2+\right.$$
$$\left.48\eta^5\mu^2+\eta^4\mu^2\left(144\mu^2-1\right)+4\eta^3\mu^2\left(24\mu^2-5\right)+8\eta^2\mu^4\left(28\mu^2-3\right)+8\eta\mu^4\left(8\mu^2-5\right)+4\mu^6\left(32\mu^2-11\right)\right)$$

Out[206]//TraditionalForm=

$$4(x-\mu)^3(\mu+x)^3\left(2x^5-x^3\left(\eta(\eta+2)+4\mu^2\right)+\mu^2x\left(\eta^2+2\mu^2\right)\right)^2$$

In[207]:=

```
𝒱ψ[Sqrt[μ^2 + η^2], η, x]
```

Out[207]=

$$1-\frac{3x^2}{\left(x^2-\mu^2\right)^2}+\frac{2(1-\eta)}{x^2-\mu^2}-\frac{-\frac{1}{4}+\eta^2+\mu^2}{x^2}$$

- $r_{\mu,\eta}$

In[208]:=

```
rμη = Collect[
  Together[Simplify[16 x^4 (x^2 - μ^2)^6 DψDown^2 (𝒱ψDown - 𝒱ψ[Sqrt[μ^2 + η^2], η, x]),
    x > μ > 0]], {x, μ}, Simplify]
```

Out[208]=

$$12x^{14}+\eta^8\mu^8+4\eta^6\mu^{10}+4\eta^4\mu^{12}+x^{12}\left(4\eta\left(-18-5\eta+4\eta^2+\eta^3\right)-44\mu^2\right)+$$
$$x^{10}\left(-\eta^2(2+\eta)^2\left(-3+8\eta+4\eta^2\right)+8\eta\left(5+6\eta-10\eta^2-3\eta^3\right)\mu^2+40\mu^4\right)+$$
$$x^8\left(\eta^4(2+\eta)^4+\eta^2\left(52+96\eta+147\eta^2+96\eta^3+20\eta^4\right)\mu^2+4\eta\left(70+2\eta+40\eta^2+15\eta^3\right)\mu^4+40\mu^6\right)+$$
$$x^6\left(-4\eta^5(2+\eta)^3\mu^2-\eta^2\left(80+116\eta+171\eta^2+144\eta^3+40\eta^4\right)\mu^4-\right.$$
$$\left.8\eta\left(45+14\eta+20\eta^2+10\eta^3\right)\mu^6-100\mu^8\right)+x^4\left(6\eta^6(2+\eta)^2\mu^4+\right.$$
$$\left.\eta^2\left(16+24\eta+81\eta^2+96\eta^3+40\eta^4\right)\mu^6+4\eta\left(20+27\eta+20\eta^2+15\eta^3\right)\mu^8+68\mu^{10}\right)+$$
$$x^2\left(-4\eta^7(2+\eta)\mu^6-4\eta^3\left(-4+3\eta+6\eta^2+5\eta^3\right)\mu^8-8\eta\left(-4+4\eta+2\eta^2+3\eta^3\right)\mu^{10}-16\mu^{12}\right)$$

In[209]:=

```
rμη // TraditionalForm
```

Out[209]//TraditionalForm=

$$\eta^8 \mu^8 + 4\eta^6 \mu^{10} + 4\eta^4 \mu^{12} + 12 x^{14} + x^{12} \left(4\eta \left(\eta^3 + 4\eta^2 - 5\eta - 18\right) - 44\mu^2\right) + x^{10} \left(-\eta^2 (\eta+2)^2 \left(4\eta^2 + 8\eta - 3\right) + 8\eta \left(-3\eta^3 - 10\eta^2 + 6\eta + 5\right)\mu^2 + 40\mu^4\right) + x^8 \left(\eta^4 (\eta+2)^4 + 4\eta \left(15\eta^3 + 40\eta^2 + 2\eta + 70\right)\mu^4 + \eta^2 \left(20\eta^4 + 96\eta^3 + 147\eta^2 + 96\eta + 52\right)\mu^2 + 40\mu^6\right) + x^6 \left(-4\eta^5 (\eta+2)^3 \mu^2 - 8\eta \left(10\eta^3 + 20\eta^2 + 14\eta + 45\right)\mu^6 - \eta^2 \left(40\eta^4 + 144\eta^3 + 171\eta^2 + 116\eta + 80\right)\mu^4 - 100\mu^8\right) + x^4 \left(6\eta^6 (\eta+2)^2 \mu^4 + 4\eta \left(15\eta^3 + 20\eta^2 + 27\eta + 20\right)\mu^8 + \eta^2 \left(40\eta^4 + 96\eta^3 + 81\eta^2 + 24\eta + 16\right)\mu^6 + 68\mu^{10}\right) + x^2 \left(-4\eta^7 (\eta+2)\mu^6 - 8\eta \left(3\eta^3 + 2\eta^2 + 4\eta - 4\right)\mu^{10} - 4\eta^3 \left(5\eta^3 + 6\eta^2 + 3\eta - 4\right)\mu^8 - 16\mu^{12}\right)$$

In[210]:=

```
rμη /. μ → 0 // Simplify
```

Out[210]=

$$x^8 \left(12 x^6 + \eta^4 (2+\eta)^4 - x^2 \eta^2 (2+\eta)^2 \left(-3 + 8\eta + 4\eta^2\right) + 4 x^4 \eta \left(-18 - 5\eta + 4\eta^2 + \eta^3\right)\right)$$

- $x^{\&}$

In[211]:=

```
xamp[ν_, η_] :=
 NSolveValues[(12 x^14 - 72 x^12 η + 12 x^10 η^2 - 20 x^12 η^2 - 20 x^10 η^3 + 16 x^12 η^3 + 16 x^8 η^4 - 45 x^10 η^4 +
      4 x^12 η^4 + 32 x^8 η^5 - 24 x^10 η^5 + 24 x^8 η^6 - 4 x^10 η^6 + 8 x^8 η^7 + x^8 η^8 - 44 x^12 μ^2 + 40 x^10 η μ^2 +
      52 x^8 η^2 μ^2 + 48 x^10 η^2 μ^2 + 96 x^8 η^3 μ^2 - 80 x^10 η^3 μ^2 + 147 x^8 η^4 μ^2 - 24 x^10 η^4 μ^2 -
      32 x^6 η^5 μ^2 + 96 x^8 η^5 μ^2 - 48 x^6 η^6 μ^2 + 20 x^8 η^6 μ^2 - 24 x^6 η^7 μ^2 - 4 x^6 η^8 μ^2 + 40 x^10 μ^4 +
      280 x^8 η μ^4 - 80 x^6 η^2 μ^4 + 8 x^8 η^2 μ^4 - 116 x^6 η^3 μ^4 + 160 x^8 η^3 μ^4 - 171 x^6 η^4 μ^4 +
      60 x^8 η^4 μ^4 - 144 x^6 η^5 μ^4 + 24 x^4 η^6 μ^4 - 40 x^6 η^6 μ^4 + 24 x^4 η^7 μ^4 + 6 x^4 η^8 μ^4 + 40 x^8 μ^6 -
      360 x^6 η μ^6 + 16 x^4 η^2 μ^6 - 112 x^6 η^2 μ^6 + 24 x^4 η^3 μ^6 - 160 x^6 η^3 μ^6 + 81 x^4 η^4 μ^6 -
      80 x^6 η^4 μ^6 + 96 x^4 η^5 μ^6 + 40 x^4 η^6 μ^6 - 8 x^2 η^7 μ^6 - 4 x^2 η^8 μ^6 - 100 x^6 μ^8 + 80 x^4 η μ^8 +
      108 x^4 η^2 μ^8 + 16 x^2 η^3 μ^8 + 80 x^4 η^3 μ^8 - 12 x^2 η^4 μ^8 + 60 x^4 η^4 μ^8 - 24 x^2 η^5 μ^8 -
      20 x^2 η^6 μ^8 + η^8 μ^8 + 68 x^4 μ^10 + 32 x^2 η μ^10 - 32 x^2 η^2 μ^10 - 16 x^2 η^3 μ^10 - 24 x^2 η^4 μ^10 +
      4 η^6 μ^10 - 16 x^2 μ^12 + 4 η^4 μ^12 /. μ → Sqrt[μ2[ν, η]]) == 0, x, Reals] // Max
zamp[ν_, η_] := ψDown1[ν, η, xamp[ν, η]];
```

- Lemma 3.4

In[213]:=

```
FullSimplify[((rμη /. {x → xhashμ[μ, η]}))] /. Sqrt[η (η (2 + η)^2 + 16 μ^2)] → ρ
```

Out[213]=

$$-\frac{3}{16}\eta^2 \left(\eta (2+\eta)^2 + 16\mu^2\right) \left(10\eta^8 + \eta^9 + 16\mu^6\rho + \eta^7 \left(40 + 4\mu^2 + \rho\right) + 4\eta^6 \left(20 + 13\mu^2 + 2\rho\right) + 32\eta \left(7\mu^6 + 3\mu^4\rho\right) + 4\eta^5 \left(20 + \mu^4 + 6\rho + \mu^2 (54+\rho)\right) + 4\eta^3 \left(4\rho + \mu^4 (96+\rho) + 8\mu^2 (7 + 3\rho)\right) + 16\eta^2 \left(3\mu^6 + 5\mu^2\rho + \mu^4 (28 + 3\rho)\right) + 4\eta^4 \left(22\mu^4 + 8(1+\rho) + \mu^2 (92 + 9\rho)\right)\right)$$

- Asymptotics of $\underset{\sim}{\psi}_{\nu,\eta}$(x)

In[214]:=

```
ψDownasympt = Series[ψDown[ν, η, x], {x, Infinity, 3}] /. μ → Sqrt[ν^2 - η^2] // Simplify
```

Out[214]=

$$x - \frac{1}{4}\pi(-1 + 2\nu) + \frac{\eta + \frac{\nu^2}{2}}{x} + \frac{-12\eta^3 - \eta^4 + 12\eta\nu^2 + \nu^4}{24 x^3} + O\left[\frac{1}{x}\right]^4$$

- Condition $\left(C_3'\right)$

In[215]:=

```
ψasympt - ψDownasympt /. μ → Sqrt[μ2[ν, η]] // Simplify
```

Out[215]=

$$\frac{3}{8\,x} + \frac{-63 - 144\,\eta - 192\,\eta^2 + 64\,\eta^3 + 16\,\eta^4 + 184\,\nu^2}{384\,x^3} + O\left[\frac{1}{x}\right]^4$$

- Computing $\psi_{\nu,\eta}(x)$ following the ideas from [Ho17]. NB: may not always work for some values of the parameters beyond those in the paper

In[216]:=

```
ψ[ν_, η_, x_] := Module[{aaa}, aaa = ArcTan[
    BesselJ[-1 + ν, x] - ((η + ν) BesselJ[ν, x])/x, BesselY[-1 + ν, x] - ((η + ν) BesselY[ν, x])/x];
   aaa - 2 π Round[1/(2 π) (aaa - If[x ≥ xhash[ν, η], Max[ψDown1[ν, η, x], 0], π/2])]
  ];
```

- Inverse function of the bound

In[217]:=

```
ψDownInv[ν_, η_, y_, prec_ : MachinePrecision] :=
  If[y > zamp[ν, η], SolveValues[ψDown1[ν, η, x] == y && x > xamp[ν, η],
     x, WorkingPrecision → prec][[1]], Null];
```

- Figure $\psi_{\nu,\eta}(x)$ bounds

In[218]:=

```
νs3 = {5, 5, 50, 200};
ηs3 = {1, 5, 25, 80};
xmaxs3 = {11, 11, 61, 217};
xamp3 = Table[xamp[νs3[[j]], ηs3[[j]]], {j, 1, 4}]
zamp3 = Table[zamp[νs3[[j]], ηs3[[j]]], {j, 1, 4}]
μs3 = Sqrt[μ2[νs3, ηs3]] // N
wp = 8;
```

Out[221]=

```
{6.62291, 5.40062, 50.4812, 196.731}
```

Out[222]=

```
{1.77658, 1.5724, 1.23801, -0.0685952}
```

Out[223]=

```
{4.89898, 0., 43.3013, 183.303}
```

In[225]:=

```
yticks = {yt = {0.5, 1, 1.5}, MaTeX[yt, Magnification → 0.8],
    Table[1, 3], Table[Directive[Thin, Dashed], 3]} // Transpose;
xt = {{6, 10}, {4, 8}, {48, 56}, {190, 210}};
xticks = Table[{xt[[j]], MaTeX[xt[[j]], Magnification → 0.8]} // Transpose, {j, 1, 4}];
Do[PrependTo[xticks[[j]],
  {xamp3[[j]], MaTeX["x^@_{\\nu,\\eta}", Magnification → 0.8], 0}], {j, 1, 4}]
```

In[229]:=

```
upzeros =
  Table[NSolveValues[{BesselJ[-1 + νs3[[j]], x] - ((ηs3[[j]] + νs3[[j]]) BesselJ[νs3[[j]], x])/x == 0,
     μs3[[j]] < x < xmaxs3[[j]]}, x, WorkingPrecision → wp], {j, 1, 4}];
PrependTo[upzeros[[2]], μs3[[2]]];
upzeros
wpzeros =
 Table[NSolveValues[{BesselY[-1 + νs3[[j]], x] - ((ηs3[[j]] + νs3[[j]]) BesselY[νs3[[j]], x])/x == 0,
    μs3[[j]] < x < xmaxs3[[j]]}, x, WorkingPrecision → wp], {j, 1, 4}]
upzerosup = Table[ψDownInv[νs3[[j]], ηs3[[j]], Pi / 2 + Pi k], {j, 1, 4}, {k, 0, 1}]
wpzerosup = Table[ψDownInv[νs3[[j]], ηs3[[j]], Pi], {j, 1, 4}]
```

Out[231]=

```
{{5.9623511, 10.396398}, {0., 9.9361095},
 {44.120394, 58.806983}, {184.41322, 213.18625}}
```

Out[232]=

```
{{8.4735224}, {7.8377378}, {55.275126}, {207.76371}}
```

Out[233]=

```
{{Null, 10.4522}, {Null, 10.0375}, {51.8426, 59.2605}, {205.511, 215.157}}
```

Out[234]=

```
{8.56122, 8.03116, 56.0603, 210.753}
```

In[235]:=

```
figcomparisonUp = GraphicsGrid[ArrayReshape[Table[
    Show[
     Plot[1 / Pi {ψ[νs3⟦j⟧, ηs3⟦j⟧, x], ψDown1[νs3⟦j⟧, ηs3⟦j⟧, x]} // Evaluate,
      {x, xamp3⟦j⟧, xmaxs3⟦j⟧}, PlotLabel → MaTeX["\\nu=" <> ToString[νs3⟦j⟧] <>
         ", \\eta=" <> ToString[ηs3⟦j⟧]], Epilog → {Black, Thin,
        Line[{{upzeros⟦j⟧⟦2⟧, 1.5}, {upzeros⟦j⟧⟦2⟧, 1.65}}],
        Line[{{wpzeros⟦j⟧⟦1⟧, 1.0}, {wpzeros⟦j⟧⟦1⟧, 1.25}}],
        Line[{{upzeros⟦j⟧⟦1⟧, 0.5}, {upzeros⟦j⟧⟦1⟧, -0.5}}],
        Line[{{xamp3⟦j⟧, 0}, {xamp3⟦j⟧, zamp3⟦j⟧ / Pi}}],
        Inset[MaTeX["u'_{\\nu,\\eta,2}\\approx" <> ToString[DecimalForm[
              upzeros⟦j⟧⟦2⟧, 6]] <> "<" <> ToString[DecimalForm[upzerosup⟦j⟧⟦2⟧, 6]],
          Magnification → 0.8], {upzeros⟦j⟧⟦2⟧, 1.65}, Scaled[{1, 0}]],
        Inset[MaTeX["w'_{\\nu,\\eta,1}\\approx" <> ToString[DecimalForm[
              wpzeros⟦j⟧⟦1⟧, 6]] <> "<" <> ToString[DecimalForm[wpzerosup⟦j⟧, 6]],
          Magnification → 0.8], {wpzeros⟦j⟧⟦1⟧, 1.25}, Scaled[{1, 0}]],
        Inset[If[j ≠ 2,
          MaTeX["u'_{\\nu,\\eta,1}\\approx" <>
            ToString[DecimalForm[upzeros⟦j⟧⟦1⟧, 6]] <> If[j < 3,
             "\\text{ no bound}",
             "<" <> ToString[DecimalForm[upzerosup⟦j⟧⟦1⟧, 6]]],
           Magnification → 0.8],
          MaTeX["u'_{\\nu,\\eta,1}=0", Magnification → 0.8]
         ], {upzeros⟦j⟧⟦1⟧, -0.5}, Scaled[{0, 1}]]
       },
      PlotRange → {{μs3⟦j⟧, xmaxs3⟦j⟧}, {-0.85, 1.85}}, Ticks → {xticks⟦j⟧, yticks}],
     Plot[1 / Pi {ψ[νs3⟦j⟧, ηs3⟦j⟧, x], ψDown1[νs3⟦j⟧, ηs3⟦j⟧, x]} // Evaluate,
      {x, μs3⟦j⟧, xamp3⟦j⟧}, PlotStyle → {Automatic, Dashed},
      PlotRange → {Automatic, {-0.7, 1.85}}]
    ], {j, 1, 4}], {2, 2}], Frame → All]
```

Out[235]=



In[236]:=

```
Export[SaveDir <> "figcomparisonUp.pdf", figcomparisonUp]
```

Out[236]=

```
./figcomparisonUp.pdf
```

§4. Benchmarking and conclusions

- Series coefficients

In[237]:=

```
A1[ν_, β_] := β; (* A_ν^(1) (β) *)
A2[ν_, β_] := β - (4 ν^2 - 1) / (8 β); (* A_ν^(2) (β) *)
A3[ν_, β_] := β - (4 ν^2 - 1) / (8 β) - 4 (4 ν^2 - 1) (28 ν^2 - 31) / (3 (8 β) ^3);
(* A_ν^(3) (β) *)
β[ν_, k_] := Pi (ν / 2 + k - 1 / 4); (* β_ν,k *)
(* A4[ν_,β_]:=β-(4ν^2-1)/(8β)-4(4ν^2-1)(28ν^2-31)/(3(8β)^3)-
   32(4ν^2-1)(1328ν^43928ν^2+3779)/(15(8β)^5); *)
```

In[241]:=

```
Series[A1[ν, β[ν, k]] + A2[ν, β[ν, k]] + A3[ν, β[ν, k]], {k, Infinity, 4}]
```

Out[241]=

$$3\pi k + \frac{3}{4}\pi(-1+2\nu) - \frac{-1+4\nu^2}{4\pi k} + \frac{(-1+2\nu)^2(1+2\nu)}{16\pi k^2} + \frac{-\frac{(-1+2\nu)^3(1+2\nu)}{64\pi} - \frac{(-1+4\nu^2)(-31+28\nu^2)}{384\pi^3}}{k^3} + \frac{(-1+2\nu)^2(1+2\nu)(-31+2\pi^2-8\pi^2\nu+28\nu^2+8\pi^2\nu^2)}{512\pi^3 k^4} + O\left[\frac{1}{k}\right]^5$$

- Previous bounds

- $\tilde{q}_{\nu,k}$

In[242]:=

```
QWUp[ν_, k_, prec_ : MachinePrecision] := Module[{az = AiryAiZero[k]},
   If[ν > 0, N[ν - az (ν / 2) ^ (1 / 3) + 3 / 20 az ^ 2 (2 / ν) ^ (1 / 3), wp], None]];
```

- $q_{\tilde{\nu},k}$

In[243]:=

```
QWDown[ν_, k_, prec_ : MachinePrecision] := Module[{az = AiryAiZero[k]},
   If[ν > 0, N[ν - az (ν / 2) ^ (1 / 3), wp], None]];
```

- $\tilde{\ell}_{\nu,k}$

In[244]:=

```
ELUp[ν_, k_] := If[ν ≤ 1 / 2,
   A2[ν, β[ν, k]],
   A3[ν, β[ν, k]]
  ];
```

- $\ell_{\tilde{\nu},k}$

In[245]:=

```
ELDown[ν_, k_] := Which[ν ≤ 1 / 2, A3[ν, β[ν, k]],
  ν < Sqrt[31 / 28], A2[ν, β[ν, k]],
  True, None]
```

- $\tilde{\ell}_{\nu,k}$

In[246]:=

```
ELprimeUp[ν_, k_, prec_ : MachinePrecision] :=
  Module[{az = AiryAiZero[k], az1 = If[k == 1, 0, AiryAiZero[k - 1]], apz, z},
   apz = Abs[z /. FindRoot[AiryAiPrime[z], {z, az, az1}, WorkingPrecision → wp]];
```

$$N\left[apz\left(\frac{8\,apz^{3/2}}{27}+\frac{\nu}{2}\right)^{1/3}+\nu+\frac{9\,apz^{2}}{10\times 2^{2/3}\left(16\,apz^{3/2}+27\,\nu\right)^{1/3}},\ wp\right]\right];$$

Appendix B

§B.1. Bounds for zeros of Bessel functions

- Parameters

In[247]:=

```
νsB1 = {0, 1 / 2, 1, 5, 10, 50, 100, 1000, 10 000, 100 000, 500 000};
ksB1 = {1, 2, 5, 10, 50, 100, 1000, 10 000, 100 000, 500 000};
wp = 16;
```

- Pretty LaTeX formatting

In[250]:=

```
MyMax[args_] := Select[args, NumberQ] // Max;
MyMin[args_] := Select[args, NumberQ] // Min;
```

In[252]:=

```
MyDecForm[x_, l_] :=
  Module[{lint}, If[NumberQ[x], lint = StringLength[ToString[Floor[x]]];
    ToString[DecimalForm[x, {l, l - lint} ]], " "]];
```

In[253]:=

```
endst = "\\\\\n";
endeverystr = "\\\\\\nopagebreak\n";
```

In[255]:=

```
MakeLineJ[k_, bd_, l_, Myf_, showdiff_ : False, dl_ : 1] :=
  Module[{str, minmax, b, stradd},
   str = " & " <> ToString[k];
   minmax = Myf[bd];
   Do[
    stradd = If[TrueQ[b == minmax],
       " & {\\color{red}" <> MyDecForm[b, l] <> "}" ,
       " & " <> MyDecForm[b, l]];
    str = str <> stradd, {b, bd}];
   If[showdiff,
    If[bd[[1]] ≠ minmax,
     str = str <> " & " <> ToString[TeXForm[ScientificForm[bd[[1]] / minmax - 1,
          dl, NumberPoint → If[dl == 1, "", "." ]]]], str = str <> " & "]
    ];
   str
  ];
```

■ Table 1 (remove semicolon to get LaTeX code)

In[256]:=

```
StringJoin[Table[StringJoin["\\multirow{" <> ToString[Length[ksB1]] <> "}{*}{$",
    ToString[TeXForm[ν]],
    "$}",
    Table[StringJoin[MakeLineJ[k, {θDownInv[ν, Pi (k - 1 / 2), wp],
        N[ELUp[ν, k], wp], QWUp[ν, k, wp]}, 15, MyMin, True, 1],
      If[k == Last[ksB1], endst <> "\\midrule\n", endeverystr]], {k, ksB1 }]
   ], {ν, νsB1}]];
```

■ Table 2 (remove semicolon to get LaTeX code)

In[257]:=

```
StringJoin[Table[StringJoin["\\multirow{" <> ToString[Length[ksB1]] <> "}{*}{$",
    ToString[TeXForm[ν]],
    "$}",
    Table[StringJoin[MakeLineJ[k, {θUpInv[ν, Pi (k - 1 / 2), wp],
        N[ELDown[ν, k], wp], QWDown[ν, k, wp]}, 15, MyMax, True, 1 ],
      If[k == Last[ksB1], endst <> "\\midrule\n", endeverystr]], {k, ksB1 }]
   ], {ν, νsB1}]];
```

■ Figure accuracy of Bessel zeros bounds

In[258]:=

```
vsB1Logs = (Range[28] - 4) / 4;
ksB1Logs = (Range[25] - 1) / 4;
wp = 16;
xyticks = {0, 2, 4, 6};
errs =
  Flatten[Table[{νlog, klog, Log[10, θDownInv[10^νlog, Pi (Floor[10^klog] - 1 / 2), wp] /
        θUpInv[10^νlog, Pi (Floor[10^klog] - 1 / 2), wp] - 1]},
    {νlog, νsB1Logs}, {klog, ksB1Logs}], 1];
(* figerrj=cleanContourPlot[ListContourPlot[errs,
   Contours→Table[c,{c,-3,-14,-1}],PlotLegends→BarLegend[Automatic,All],
   Frame →{{True,False},{True,False}},
   FrameLabel→MaTeX[{"\\log_{10}\\nu","\\log_{10} k"}],
   FrameTicks→{{xyticks, MaTeX[xyticks]}//Transpose,
     {xyticks, MaTeX[xyticks]}//Transpose}]] *)
figerrj = Labeled[cleanContourPlot[ListContourPlot[errs,
    Contours → Table[c, {c, -3, -14, -1}], PlotLegends → BarLegend[Automatic, All],
    Frame → {{True, False}, {True, False}},
    FrameTicks → {{xyticks, MaTeX[xyticks]} // Transpose,
      {xyticks, MaTeX[xyticks]} // Transpose}]],
  MaTeX[{"\\log_{10}\\nu", "\\log_{10} k"}],
  {Bottom, Left}, RotateLabel → True, Spacings → {0, -0.7}]
```

Out[263]=



In[264]:=

```
Export[SaveDir <> "figerrj.pdf", figerrj]
```

Out[264]=

```
./figerrj.pdf
```

§B.2. Bounds for zeros of derivatives of Bessel functions

■ Pretty LATEX formatting

In[265]:=
```
MakeLineJprime[k_, outbd_, bd_, l_, Myf_, showdiff_ : False, dl_ : 1] :=
  Module[{str, minmax, b, stradd},
   str = " & " <> ToString[k] <> " & " <> MyDecForm[outbd, l];
   minmax = Myf[bd];
   Do[
    stradd = If[TrueQ[b == minmax],
       " & {\\color{red}" <> MyDecForm[b, l] <> "}" ,
       " & " <> MyDecForm[b, l]];
    str = str <> stradd, {b, bd}];
   If[showdiff,
    If[bd[[1]] ≠ minmax,
     str = str <> " & " <> ToString[TeXForm[ScientificForm[bd[[1]] / minmax - 1,
          dl, NumberPoint → If[dl == 1, "", "." ]]]], str = str <> " & "]
    ];
   str
  ];
```

■ Table 3 (remove semicolon to get LATEX code)

In[266]:=
```
νstar0 = ν /. FindRoot[zstar[ν] / Pi == 1 / 2, {ν, 1}];
```

In[267]:=
```
StringJoin[Table[StringJoin["\\multirow{" <> ToString[Length[ksB1]] <> "}{*}{$",
    ToString[TeXForm[ν]],
    "$}",
    Table[StringJoin[
      MakeLineJprime[k, If[ν > νstar0 || k > 1, ϕUpInv[ν, Pi (k - 1 / 2), wp], None],
       {ϕDownInv[ν, Pi (k - 1 / 2), wp], ELprimeUp[ν, k, wp]}, 15, MyMin, True, 1],
      If[k == Last[ksB1], endst <> "\\midrule\n", endeverystr]], {k, ksB1 }]
   ], {ν, νsB1}]];
(* remove semicolon to get LaTEX for Table 3*)
```

■ Figure accuracy of Bessel derivative zeros bounds

In[268]:=

```
errsprime = Flatten[Table[If[10^νlog > νstar0 || Floor[10^klog] > 1,
      {νlog, klog, Log[10, θDownInv[10^νlog, Pi (Floor[10^klog] - 1/2), wp] /
          θUpInv[10^νlog, Pi (Floor[10^klog] - 1/2), wp] - 1]}],
     {νlog, νsB1Logs}, {klog, ksB1Logs}], 1];
errsprime = Select[errsprime, ! TrueQ[# == Null] &];
(* figerrjprime=cleanContourPlot[ListContourPlot[errsprime,
   Contours→Table[c,{c,-3,-14,-1}],PlotLegends→BarLegend[Automatic,All],
   Frame →{{True,False},{True,False}},
   FrameLabel→MaTeX[{"\\log_{10}\\nu","\\log_{10} k"}],
   FrameTicks→{{xyticks, MaTeX[xyticks]}//Transpose,
     {xyticks, MaTeX[xyticks]}//Transpose}]] *)
figerrjprime = Labeled[cleanContourPlot[ListContourPlot[errsprime,
     Contours → Table[c, {c, -3, -14, -1}], PlotLegends → BarLegend[Automatic, All],
     Frame → {{True, False}, {True, False}},
     FrameTicks → {{xyticks, MaTeX[xyticks]} // Transpose,
       {xyticks, MaTeX[xyticks]} // Transpose}]],
  MaTeX[{"\\log_{10}\\nu", "\\log_{10} k"}], {Bottom, Left},
  RotateLabel → True, Spacings → {0, -0.7}]
```

Out[270]=



In[271]:=

```
Export[SaveDir <> "figerrjprime.pdf", figerrjprime]
```

Out[271]=

```
./figerrjprime.pdf
```